

ProTem

[Eric Hehner](#)

ProTem is a programming system that serves as both programming language and operating system, and includes a theorem prover to check each step of program composition. This document is an informal specification of ProTem. Formal specifications of the data types and program semantics can be found in the book [a Practical Theory of Programming](#) (with syntactic differences). “ProTem” also means “for now”, short for the Latin words “pro tempore”.

Programming languages and operating system languages have a lot of functionality in common, but differ greatly in syntax and terminology. These differences are historical, accidental, and unnecessary. They complicate a programmer's life with no benefit. For example, a file is just a variable; file update and storage are just assignment. By unifying the programming language and the operating system commands, both gain in functionality. Communication channels and file piping are as useful in programming as they are in operating systems. Directories are useful in large-scale multi-programmer programs. Conditional execution (**if**) and indexed loops (**for**) are useful operating system commands.

ProTem is also designed for easy proof of correctness, including functionality, time requirements, and space requirements. To that end, loops can be constructed by labeling any block of code with a specification, and then using the specification within the block of code. For example,

$$\ll n \geq 0 \Rightarrow n' = 0 \gg \ll \mathbf{if} \ n > 0 \ \ll n := n - 1. \ \ll n \geq 0 \Rightarrow n' = 0 \gg \ll \mathbf{end} \gg \ll \mathbf{end} \gg$$

The proof methods are the subject of the book [a Practical Theory of Programming](#) and paper [Specified Blocks](#). They do not require preconditions, postconditions, invariants, or variants. If proof is not wanted, then an ordinary name can be used as label. For example,

$$\mathit{loop} \ll \mathbf{if} \ n > 0 \ \ll n := n - 1. \ \mathit{loop} \ll \mathbf{end} \gg$$

A primary design criterion is to make ProTem a small, easy-to-learn, easy-to-use language. The size of a language can be measured by the number of symbols and by the complexity of grammar structure, which can be measured by the number of nonterminals. ProTem has 8 keywords. (Python has 35, C has 36, Pascal has 36, Haskell has 37, Ada has 62, MS Basic has 205.) ProTem is presented by a Presentation Grammar, which has just the structure that a programmer needs to know, not all the structure that a compiler needs. It has 2 nonterminals (program and data) plus some informally defined kinds of names. (There is also an LL(1) grammar with 25 nonterminals and an LR(0) grammar with 14 nonterminals at the end of this document. For comparison, the Haskell grammar has 68 nonterminals, and the Python grammar has 87 nonterminals.) The design ethos demands an extremely good reason for adding a new feature to ProTem that requires a new keyword or syntax. That same design ethos will not tolerate any addition to the 2 nonterminals in the Presentation Grammar.

To judge ease of use, you need to use the language, but you may get a sense of the ease of use (and of the beauty of the language, if that is of interest) from reading example programs. For that purpose, there are example programs near the end of this document.

The design of ProTem is complete except for the following. I need to describe and compose picture and sound elements. I need to define touchpad and touchscreen gestures. I may need to define regions of documents to be clickable links.

An [implementation](#) of ProTem, written in ProTem, is partially complete.

[Contents](#)

[Introduction](#)

[Contents](#)

[Symbols](#)

[Presentation Grammar](#)

[Order of Execution and Evaluation](#)

[Using ProTem](#)

[Data](#)

[Numbers](#)

[Characters](#)

[Binary Values](#)

[Bunches](#)

[Sets](#)

[Strings](#)

[Lists](#)

[Conditional Data](#)

[Functions](#)

[Named-Data](#)

[value-Data](#)

[Quote and Unquote](#)

[Scope](#)

[Programs](#)

[Variable Definition](#)

[Assignment](#)

[Constant Definition](#)

[Data Definition](#)

[Data Recursion](#)

[Named-Data versus Data Definition](#)

[Constant Definition versus Data Definition](#)

[Sequential Composition](#)

[Concurrent Composition](#)

[if-Program](#)

[case-Program](#)

[for-Program](#)

[Program Definition](#)

[Named-Program](#)

[Named-Program versus Program Definition](#)

[Output and Input](#)

[Channel Definition](#)

[Plan](#)

[Dictionary Definition](#)

[Measuring Unit Definition](#)

[Forward Definition](#)

[Name Removal](#)

[Synonym Definition](#)

[Format](#)

[Commands](#)

[Pause](#)

[Abort](#)

[Verify](#)

[Type Checking](#)

[Session](#)

[Undo](#)

[Edit](#)

[Display](#)

[Names](#)

[Memo](#)

[Context](#)

[Predefined Names](#)

[Miscellaneous](#)

[Intentionally Omitted Features](#)

[Implementation Philosophy](#)

[Example Programs](#)

[Portation Simulation](#)

[Quote Notation Lengths](#)

[Minimum Redundancy Codes](#)

[Read Password](#)

[Grammars](#)

[LL\(1\) Grammar](#)

[LR\(0\) Grammar](#)

[Acknowledgements](#)

As you read this document the first time, you will encounter language constructs that have not been defined yet. This document has been written to minimize the problem, but due to the highly mutually recursive nature of the language constructs, there is no linear order that completely avoids the problem. (The same was true of the ALGOL-60 report, and every decent programming language document since.) When you encounter a not-yet-explained language construct, understand whatever you can about it, but do not be stopped by it. During the second and subsequent readings of this document, all language constructs fall into place.

I am perfectly well aware that “data” is a Latin word; it is the plural gerund from the verb “dare”, which means “to give”, so the data are the givens. The singular is “datum”. But this is an English document, and I have decided to use the word “data” for both singular and plural. I have also decided to say “indexes” rather than “indices”.

Symbols

ProTem has 8 keywords, plus 4 kinds of lexeme, and 71 other symbols; altogether they are:

case else for if new old plan value

number text name comment

“ ” « » ` ‘ , ,.. ; ; ; ; .. . ! ? ?? : :: := = ≠ < > ≤ ≥ _ # #1 \ | || ∞ & % † ‡

+ − × / → ↔ ∧ ∨ ^ ^^ @ * ~ † ‡ \$ ∈ □ ◁ ▷ ≡ ≐ () { } [] ⟨ ⟩ () [] ()

Keywords and multicharacter symbols cannot have spaces between the characters.

Some of the ProTem symbols may not be on your keyboard. Here are the substitutes.

for “ use "	for ” use "	for « use <<	for » use >>
for ‘ use '	for ≠ use /=	for ≤ use <=	for ≥ use >=
for − use -	for × use ><	for → use ->	for ↔ use <>
for ∧ use / \	for ∨ use \ /	for † use //	for ∈ use :~
for ◁ use <l	for ▷ use >l	for (use (:	for) use :)
for ≡ use =	for ≐ use =	for ⟨ use <:	for ⟩ use :>
for (use (l	for) use l)	for [use [l	for] use l]
for □ use []	for ∞ use <i>infinity</i>	for † use <i>true</i>	for ‡ use <i>false</i>

The names *infinity* , *true* , and *false* are predefined, and redefinable.

A number is formed as one or more decimal digits, with an optional decimal point between digits. A decimal point must have at least one digit on each side of it. Commas and spaces between digits are not allowed. Here are four examples.

0 275 27.5 0.21

A text begins with a left-double-quote, continues with any number of any characters, and ends with a right-double-quote. Within a text, a double-quote (left or right) must be underlined or written twice. Characters within a text are not limited to any alphabet. Here are six examples.

“” “abc” “don't” “Just say “no_.”” “Just say ““no””.” “♠♣♥♦”

A name is either simple or compound. A simple name is either plain or fancy. A plain simple name begins with a letter from an alphabet (this document uses the 26 small italic and 26 capital italic letters of the English alphabet), and continues with any number of letters and decimal digits, except that keywords cannot be names. A fancy simple name begins with « , continues with any number of any characters, and ends with » . Within a fancy simple name, a « or » must be underlined. Characters within a fancy simple name are not limited to any alphabet. A compound name is composed of two or more simple names joined with backslash \ characters. For examples:

plain simple names: *x A123 refStack*

fancy simple names: «Rick & Margaret» «*x' ≥ x*» «left<center>right»

compound names: *ProTem\grammars\LL1 CS\«grad recruiting»\«2016-9-8»*

At each point in a program, a name is one of

newname: a simple name that is not defined in the current scope,
or a compound name that is not defined in its dictionary in the current scope.

oldname: a simple name that is defined in the current scope,
or a compound name that is defined in its dictionary in the current scope.

An oldname is defined as one of: *variablename*, *constantname*, *dataname*,
programname, *channelname*, *unitname*, or *dictionaryname*.

A comment begins with ` and ends at the end of the line. Characters within a comment are not limited to any alphabet. For example: ` I♥ProTem

Presentation Grammar

ProTem will be explained in complete detail in the following sections. In this section, for reference, we present the grammar. There are 36 ways of forming a program, all listed in the left column. There are a few words of explanation in the right column.

new newname : data := data	define variablename with type and initial value
new newname := data	define constantname and evaluate data
new newname (data)	define dataname but do not evaluate data
new newname [program]	define programname but do not execute program
new newname ? data ! data	define channelname with type and initial value
new newname #1	define measuring unitname
new newname \	define dictionaryname
new newname \\ dictionaryname	define dictionaryname with definitions
new newname oldname	define synonym
new newname	forward definition of dataname or programname
old oldname	undefine, remove, or hide
variablename := data	assign variable to value
channelname ! data	to channel send output
channelname ? data (data) data	from channel receive input in this pattern
channelname ? data (data) data ! channelname	from channel receive input in this pattern and echo
channelname ? ! channelname	from channel receive input in default pattern and echo
simplename [program]	define programname and execute program
programname	execute (call) named-program
program . program	sequential composition
program program	concurrent composition
if data [program]	if -program
if data [program] else [program]	if-else -program
case data [program]	case -program
case data [program] else [program]	case-else -program
program] [program	cases in a case -program or case-else -program
for simplename : data [program]	for -program, index is constantname
plan simplename : data [program]	plan, parameter is constantname
plan simplename := data [program]	plan, parameter is variablename
plan simplename ! data [program]	plan, parameter is output channelname
plan simplename ? data [program]	plan, parameter is input channelname
plan simplename \ [program]	plan, parameter is dictionaryname
program data	plan, data argument
program variablename	plan, variable argument
program channelname	plan, channel argument
program dictionaryname	plan, dictionary argument
[program]	scope, program brackets

There are 60 ways of expressing data. Examples and pronunciations are shown on the right side.

number	0 1.2
∞	infinity, the infinite number
data & data	complex number $x&y = x + iy$
data %	percentage $x\% = x/100$
+ data	plus, identity
- data	minus, negation, not
data + data	plus, addition

data – data	minus, subtraction
data × data	times, multiplication
data / data	divided by, division
data ^ data	to the power, exponentiation
data ^^ data	scale, scientific notation $x^{^y} = x \times 10^{^y}$
⊤	top, true
⊥	bottom, false
data ∧ data	minimum, conjunction, and, set intersection
data ∨ data	maximum, disjunction, or, set union
data = data	equals, equation
data ≠ data	not equals, differs from, exclusive or
data < data	less than, strict implication, strict subset
data > data	greater than, strict reverse implication, strict superset
data ≤ data	less than or equal to, implication, subset
data ≥ data	greater than or equal to, reverse implication, superset
data , data	bunch union $(0, 2), (2, 5) = 0, 2, 5$
data ... data	bunch from (including) to (excluding) $0,..3 = 0,1,2$
data ‘ data	bunch intersection $(0, 2, 5) ‘ (2, 5, 9) = 2, 5$
data : data	bunch inclusion $2, 5: 0, 2, 5, 9$
data :: data	reverse bunch inclusion $0, 2, 5, 9:: 2, 5$
ϕ data	bunch size, bunch cardinality $ϕ(0,..5) = 5$
{ data }	set $\{0, 2, 5\}$ $\{0,..5\}$ set brackets
~ data	contents of a set or list $\sim\{0, 2, 5\} = 0, 2, 5$
\$ data	set size, set cardinality $\$\{0, 2, 5\} = 3$
data ∈ data	elements of a set $2, 3 ∈ \{0, 2, 3, 5\}$
∄ data	power $∄(0, 1) = \{null\}, \{0\}, \{1\}, \{0, 1\}$
text	“” “abc” “Just say “no.””
data ; data	string join $0; 2; 5$
data ;.. data	string from (including) to (excluding) $0;..3 = 0;1;2$
data _ data	string indexing $(2; 3; 7; 3)_2 = 7$
data < data > data	string modification $3; 4; 5 <1> 6 = 3; 6; 5$
↔ data	string length $↔(2; 3; 7; 3) = 4$
data * data	definite repetition $3*2 = 2; 2; 2$
* data	indefinite repetition $*2 = nat*2$
[data]	list $[3; 6; 5]$ list brackets
data ;; data	list join $[3; 6];:[7; 4] = [3; 6; 7; 4]$
# data	list length, function size $\#[2; 3; 7; 3] = 4$
data data	list index, function argument, composition $f x$
data @ data	pointer indexing $L@(i; j) = L i j$
⟨ simplename : data . data ⟩	function, parameter is constantname $\langle n: nat. n+1 \rangle$
data → data	function, function space $nat \rightarrow bin = \langle n: nat. bin \rangle$
□ data	domain of a list or function $\square \langle n: nat. n+1 \rangle = nat$
data data	selective union $2 \rightarrow 8 [3; 6; 7; 4] = [3; 6; 8; 4]$
variablename	variable name
constantname	constant name
dataname	data name and evaluate data
channelname ?	the most recent data read on the channel
channelname ??	test for written but unread data on the channel
unitname	unit name, positive finite real number constant
data ≡ data ⇒ data	conditional data, if data then data else data
simplename (data)	named-data, defines dataname, data brackets
value simplename : data := data [program]	value -data, defines variablename
(data)	evaluation brackets

Order of Execution and Evaluation

Here is the order of execution of the forms of program.

0	new old := ! ? programname plan if case for []	
1	plan argument	left-to-right
2		
3	.	

Program brackets [] can always be used to change the order of execution.

Here is the order of evaluation of the forms of data.

0	number text name $\top \perp \infty$ () [] { } < > () (;) value	
1	list index function argument postfix % ? ?? infix _ @ &	left-to-right
2	prefix + - ¢ \$ \leftrightarrow # ~ / \square * infix * \rightarrow ^ ^^	right-to-left
3	infix \times / \wedge \vee	left-to-right
4	infix + - ; ; ; .. ‘	left-to-right
5	infix , ,.. $\triangleleft \triangleright$	left-to-right
6	infix = \neq < > \leq \geq : :: \in	continuing
7	infix \models \Rightarrow	right-to-left

Evaluation brackets () can always be used to change the order of evaluation.

On level 6, the operators are “continuing”; this means, for example, that $a=b=c$ neither associates to the left $(a=b)=c$ nor associates to the right $a=(b=c)$, but means $(a=b)\wedge(b=c)$. Similarly $a<b=c<d$ means $(a<b)\wedge(b=c)\wedge(c<d)$.

Using ProTem

Following the prompt \diamond , key in a program. As you do so, keywords become bold, plain names become italic, and keyboard substitutes become the proper symbols. A program may stretch over many lines, including spaces, tabs, and new line characters between symbols. You may make changes to a program using the delete (backspace) key and standard cut/copy/paste editing commands. When you are finished, send the *end* character, either by pressing the escape key, or by pressing $\boxed{\text{ctl}}$, which is the simultaneous combination of the control key and the period (point) key. Then the program is checked for errors; if there are errors, you are told what the errors are, and invited to correct them; if there are no errors, your program is executed. For example, following the prompt \diamond , you can key in

```
! 2+2
```

The **!** means output to the screen (see [Output and Input](#)). Then send *end*. Since there are no errors, the program is executed, causing 4 to be printed on the screen. Thus ProTem can be used as a calculator.

As we will see in [Constant Definition](#), the program

```
new temp:= 2+2
```

followed by *end*, saves the result of the calculation $2+2$ under the name *temp*, perhaps for use in further calculation. As we will see in [Program Definition](#), the program

```
new myprogram [! “2+2=”; temp]
```

followed by *end*, defines and saves a program named *myprogram*. The saved program is not immediately executed. To execute this saved program, key in

```
myprogram
```

followed by *end*. Saved definitions can be edited using the command $\boxed{\text{ctl e}}$ (see [Edit](#)).

Data

The basic data are numbers, characters, and binary values. The data structures are bunches, sets, strings, and lists. Also, there are conditional data, functions, named-data, and **value**-data.

Numbers

Numbers are not divided into disjoint types. A natural number is an integer number; an integer number is a rational number; a rational number is a real number; a real number is a complex number. There is also an infinite number ∞ , greater than all other numbers, and $-\infty$, less than all other numbers, included in the reals.

The one-operand postfix operator $\%$ means division by 100; for examples, 99.9% , $x\%$, and $(x+y)\%$. There are two one-operand prefix operators $+$ and $-$. There are nine two-operand infix operators $+ - \times / \wedge \wedge \wedge \vee \&$. Division of integers, such as $1/2$, may produce a noninteger. Exponentiation is 2-operand infix \wedge ; for example, 1.2×10^3 (one point two times ten to the power three), which can be written more briefly as $1.2 \wedge 3$, and in general, $x \wedge y = x \times 10^y$. The operator \wedge is minimum (arms down, does not hold water; note that \wedge and \wedge are different). The operator \vee is maximum (arms up, holds water). In addition to the number symbols, there are predefined names of numbers such as pi (the ratio of a circle's circumference to its diameter), e (the base of the natural logarithms), and i (the imaginary unit, a square root of -1). The complex number $x + iy$ can be written more briefly as $x \& y$. There are predefined function names such as abs , arc , $arccos$, $arcsin$, $arctan$, $ceil$, cos , $cosh$, div , exp , $floor$, im , lb , ln , log , mod , $randInt$, $randReal$, re , $round$, sin , $sinh$, $sqrt$, tan , and $tanh$ (see [Predefined Names](#)). Predefined names can be redefined.

Characters

A character is a text of length 1. We leave it to each implementation to list the characters, and to state their order. In addition to the character symbols such as “a” (small a) and “ ” (space), there are some predefined character names: $delete$ (backspace), tab , nl (new line, next line, return, enter), $click$, $doubleclick$, and end . The end character is invisible and takes no space; it is sometimes useful as a sentinel at the end of a text. Predefined functions suc and pre give the successor and predecessor in the character order; “ ” (space) comes first and end comes last. Predefined functions $charnat$ and $natchar$ map between characters and their (possibly extended ASCII or unicode) numeric encodings. Some key combinations, such as $\boxed{ctl .}$ and $\boxed{ctl e}$, are characters and have numeric encodings.

Binary Values

The two binary values are \top and \perp . Negation is $-$, conjunction (minimum) is \wedge , disjunction (maximum) is \vee . The infix two-operand operators $=$ and \neq apply to all data in ProTem with a binary result; the two operands may even be of different types.

The order operators $< > \leq \geq$ apply to real numbers (including rationals, integers, and naturals), to characters, to binary values, to sets (subset, superset), to strings of ordered items lexicographically, and to lists of ordered items lexicographically, with a binary result. In the binary order, \perp is below \top , so \leq is implication.

The postfix operator $??$ applies to channels, and has a binary result saying whether there is written but unread data on the channel.

union operator, traditionally \cup , is \vee in ProTem. The intersection operator, traditionally \cap , is \wedge . Subset, traditionally \subseteq , is \leq ; strict subset is $<$; superset is \geq ; strict superset is $>$. The power operator $\$$ takes a bunch as operand and produces the bunch of all sets that contain only elements of the operand. For examples, $\$(0, 1) = \{null\}, \{0\}, \{1\}, \{0, 1\}$, and $\$bin = \{null\}, \{\top\}, \{\perp\}, \{bin\}$. In $\$(0, 1)$, the evaluation brackets $()$ are needed due to the order of data evaluation.

Strings

There is a predefined string name:

nil the empty string

Any number, character, binary value, set, list, and function is a one-item string, or synonymously, an item. For example, the number 2 is a one-item string, or item.

String join is denoted by a semi-colon:

$S; T$ S join T

For example,

2; 3; 5; 7

is a string of four integers. There is also the notation

$x;..y$ x to y (same pronunciation as $x;..y$)

where x and y are integers or ∞ or characters that satisfy $x \leq y$. Again, x is included and y is excluded. For examples, $0;..3 = 0;1;2$ and $5;..5 = nil$.

The length of a string is obtained by the \leftrightarrow operator. For examples, $\leftrightarrow nil = 0$, $\leftrightarrow 2 = 1$, $\leftrightarrow(2; 3; 5; 7) = 4$, and $\leftrightarrow(x;..y) = y - x$. The evaluation brackets $()$ are needed for the order of data evaluation.

A string is indexed by the $_$ operator. Indexing is from 0. For example, $(2; 3; 5; 7)_2 = 5$. A string can be indexed by a string. For example, $(3; 5; 7; 9)_{(2; 1; 2)} = 7; 5; 7$.

If S is a string and n is an index of S and i is any item, then $S \triangleleft n \triangleright i$ is a string like S except that item n is i . For example, $3; 5; 9 \triangleleft 2 \triangleright 8 = 3; 5; 8$. This operator associates from left to right, so $3; 5; 9 \triangleleft 2 \triangleright 8 \triangleleft 1 \triangleright 7 = ((3; 5; 9) \triangleleft 2 \triangleright 8) \triangleleft 1 \triangleright 7 = (3; 5; 8) \triangleleft 1 \triangleright 7 = 3; 7; 8$. And $3; 5; 9 \triangleleft 2 \triangleright 8 \triangleleft 2 \triangleright 7 = ((3; 5; 9) \triangleleft 2 \triangleright 8) \triangleleft 2 \triangleright 7 = (3; 5; 8) \triangleleft 2 \triangleright 7 = 3; 5; 7$.

A text is a more convenient notation for a string of characters.

“abc” = “a”; “b”; “c”

“He said “Hi”.” = “H”; “e”; “ ”; “s”; “a”; “i”; “d”; “ ”; “_”; “H”; “i”; “_”; “.”

“abcdefghij”_(3;..6) = “def”

Strings are equal if and only if they have the same length, and corresponding items are equal. They are ordered lexicographically. For examples,

$3; 5 < 3; 5; 2 < 3; 6$

$(3; 5) \vee (3; 5; 2) = 3; 5; 2$ maximum

A bunch of items is an item. Join distributes over bunch union, so

$(3, 4); (5, 6) = 3; 5, 3; 6, 4; 5, 4; 6$

A string is an element (elementary bunch) if and only if all its items are elements.

If S is a string and n is a natural number, then

$n * S$ n copies of S , or $n S$'s
 is a string, and
 $* S$ strings of S , or any number of S 's

is a bunch of strings. For examples,

$$3*5 = 5;5;5$$

$$3*(4, 5) = 4;4;4, 4;4;5, 4;5;4, 4;5;5, 5;4;4, 5;4;5, 5;5;4, 5;5;5$$

$$*5 = nil, 5, 5;5, 5;5;5, 5;5;5;5, \text{ and so on}$$

The $*$ operator distributes over bunch union in its left operand only.

$$null*5 = null$$

$$(2,3)*5 = 2*5, 3*5 = 5;5, 5;5;5$$

Using this semi-distributivity, we have

$$*a = nat*a$$

Lists

A list is a packaged string. It can be written as a string enclosed in list brackets. For example,

[0; 1; 2]

Let L and M be lists, let n be a natural number, and let p be a string of natural numbers. The list operators are:

$\square L$	domain of L
$\sim L$	content of L
$\# L$	length of L
$L n$	L at n , L at index n
$L @ p$	L at p , L at pointer p
$L ;; M$	L join M
LM	L composed with M
$L M$	L otherwise M , the selective union of L and M
$i \rightarrow x L$	index i is item x and otherwise L

plus the operators $L \wedge M$, $L \vee M$, $L = M$, $L \neq M$, $L < M$, $L > M$, $L \leq M$, $L \geq M$. For examples,

$$\square [10; 11; 12] = 0, 1, 2 \quad \text{the domain of a list}$$

$$\sim [10; 11; 12] = 10; 11; 12 \quad \text{the content of a list}$$

$$\# [10; 11; 12] = 3 \quad \text{the length of, or number of items in, a list}$$

$$[10; ..20] 5 = 15 \quad \text{indexing starts at zero}$$

$$[[2; 3]; 4; [5; [6; 7]]] @ (2; 1; 0) = 6$$

$$[0; ..10]; [10; ..20] = [0; ..20] \quad \text{joining lists}$$

$$[10; ..20] [3; 6; 5] = [13; 16; 15] \quad \text{composition } (LM)n = L(Mn)$$

By using the $@$ operator, a string acts as a pointer to select an item from within an irregular structure. If the list $L | M$ is indexed with n , the result is either $L n$ or $M n$ depending on whether n is in the domain $0, ..\#L$ of L . If it is, the result is $L n$, otherwise the result is $M n$.

$$[10; 11] | [0; ..10] = [10; 11; (2; ..10)]$$

$$1 \rightarrow 21 | [10; 11; 12] = [10; 21; 12]$$

The index can be a string, as in

$$(0;1) \rightarrow 6 | [[0; 1; 2]; [3; 4; 5]] = [[0; 6; 2]; [3; 4; 5]]$$

When a string or list is indexed by a structure, the result has the same structure as the index.

$$(10; ..20) _ [2; (3, 4); [5; [6; 7]]] = [12; (13, 14); [15; [16; 17]]]$$

$$[10; ..20] [2; (3, 4); [5; [6; 7]]] = [12; (13, 14); [15; [16; 17]]]$$

Let $S = 10; 11; 12$. Then

$$\begin{aligned}
& S_0(0, \{1, [2; 1]; 0\}) \\
= & S_0, \{S_1, [S_2; S_1]; S_0\} \\
= & 10, \{11, [12; 11]; 10\}
\end{aligned}$$

Let $L = [10; 11; 12]$. Then

$$\begin{aligned}
& L(0, \{1, [2; 1]; 0\}) \\
= & L_0, \{L_1, [L_2; L_1]; L_0\} \\
= & 10, \{11, [12; 11]; 10\}
\end{aligned}$$

Lists are equal if and only if they have the same length and corresponding items are equal. They are ordered lexicographically. For examples,

$$\begin{aligned}
& [3; 5] < [3; 5; 2] < [3; 6] \\
& [3; 5] \vee [3; 5; 2] = [3; 5; 2] \quad \text{maximum}
\end{aligned}$$

The list brackets $[]$ distribute over bunch union. For example,

$$[0, 1] = [0], [1]$$

Thus $[10^*nat]$ is all lists of length 10 whose items are natural, and $[4^*[6^*real]]$ is all 4 by 6 arrays of reals.

Conditional Data

The 3-operand expression $x \vDash y \vDash z$, pronounced “if x then y else z ”, has binary operand x , but y and z are of arbitrary type. For example,

$$y \neq 0 \vDash x/y \vDash \text{“nan”}$$

If $y \neq 0$ has value \top , then this data expression has number value x/y . If $y \neq 0$ has value \perp , then this data expression has text value “nan” . This operator associates from right to left. For example,

$$(a \vDash b \vDash c \vDash d \vDash e) = (a \vDash b \vDash (c \vDash d \vDash e))$$

If a has value \top , then this expression has value b , with no need to evaluate c , d , or e . If a has value \perp , there is no need to evaluate b .

Functions

A function defines a parameter; that is its only job. Let p (parameter) be any simple name, let D (domain) be any data expression (but not using p), and let B (body) be any data expression (possibly using p as a constant name for an element of D). Then $\langle p: D. B \rangle$ is a function with parameter p , domain D , and body B . For example,

$$\langle n: nat. n+1 \rangle \quad \text{map } n \text{ in } nat \text{ to } n+1$$

is the successor function on the natural numbers. The parameter name begins its scope at the left function bracket \langle and ends its scope at the right function bracket \rangle (see [Scope](#)). Consequently, the parameter name can be any simple name, even one that has already been defined in the scope that encloses the function. The \square operator gives the domain of a function. For example,

$$\square \langle n: nat. n+1 \rangle = nat$$

The $\#$ operator gives the size of the function, which is the size of its domain. For examples,

$$\# \langle n: 0..10. n+1 \rangle = 10$$

$$\# \langle n: nat. n+1 \rangle = \infty$$

A function of $n+1$ parameters is a function of 1 parameter whose body is a function of n parameters. For example, the average function

$$\langle x: rat. \langle y: rat. (x+y)/2 \rangle \rangle$$

has two parameters. The notation for applying a function to an argument is the same as that for indexing a list: adjacency. If f is a function of two parameters, then $f x y$ applies f to x and y .

Caution: in some languages, applying f to x and y is $f(x, y)$. In ProTem, comma is bunch union, and function application distributes over bunch union. So in ProTem, $f(x, y) = f x, f y$.

The predefined function *realtext* has four parameters. The first three parameters say how to format a number, and the last is the number to be formatted. For example, *realtext* 4 1 10 *pi* = “3.1416⁰”. We can define a new function

new *myrealtext* (*realtext* 4 1 10)

by supplying just three parameters, and then apply it to a number to be formatted:

myrealtext *pi* = “3.1416⁰”

When the body of a function does not use its parameter, there is a syntax that omits the unused name. For example, $2 \rightarrow 3$ means $\langle n: 2. 3 \rangle$ or choose any other parameter name. And $nat \rightarrow bin$ means $\langle n: nat. bin \rangle$ or choose any other parameter name.

Argumentation comes before bunch union in the order of data evaluation, and so it distributes over bunch union.

$(f, g)(x, y) = f x, f y, g x, g y$

If you want to apply a function to a bunch without distributing over the elements of the bunch, you must package the bunch as a set.

Allowing the body of a function to be a bunch generalizes the function to a relation. For example, $nat \rightarrow bin$ can be viewed in either of the following two ways: it is a function (with unused and omitted parameter) that maps each natural to *bin*; it is all functions with domain at least *nat* and range at most *bin*. As an example of the latter view, we have

$\langle i: int. i < 10 \rangle : nat \rightarrow bin$

If f and g are functions, then

$f | g$ “ f otherwise g ”, “the selective union of f and g ”

is a function that behaves like f when applied to an argument in the domain of f , and otherwise behaves like g .

$\square(f | g) = \square f, \square g$

$(f | g) x = (x: \square f \models f x \Rightarrow g x)$

The function $\langle f: nat \rightarrow rat. f 2 \rangle$ is “higher order”, which means it has a function-valued parameter. It can be applied to any function with domain at least *nat* and range at most *rat*.

Let f and g be functions such that $\neg(f: \square g)$ (f is not in the domain of g). Then $g f$ is the composition of g and f .

$(x: \square(g f)) = (x: \square f) \wedge (f x: \square g)$

$(g f) x = g(f x)$

Named-Data

Named-data has the form

simplename (data)

Within the data brackets (), the *simplename* stands for the data recursively. The *simplename* begins its scope at the left data bracket and ends its scope at the corresponding right data bracket (see [Scope](#)). Consequently, the name can be any simple name, even one that has already been defined in the scope that encloses the named-data.

For example, here is a bunch of texts (a pattern or grammar).

$$\text{term} \langle \text{"a"}, \text{"b"}, \text{term}; \text{"+"}; \text{term}, \text{term}; \text{"-"}; \text{term} \rangle$$

This bunch includes the text “a+b+a-a” and many more texts. It is equivalent to

$$\langle \text{"a"}, \text{"b"} \rangle; *(\langle \text{"+"}, \text{"-"} \rangle; (\langle \text{"a"}, \text{"b"} \rangle))$$

Data is evaluated as needed. The preceding bunch is infinite, but it may appear in a context that does not require its complete evaluation. For example, the binary expression

$$\text{"a+b+a-a"}: \text{term} \langle \text{"a"}, \text{"b"}, \text{term}; \text{"+"}; \text{term}, \text{term}; \text{"-"}; \text{term} \rangle$$

can be evaluated to \top without fully evaluating the named-data to the right of the colon.

Here is the factorial function.

$$\text{fact} \langle 0 \rightarrow 1 \mid \langle n: \text{nat}+1. n \times \text{fact} (n-1) \rangle \rangle$$

If we were to fully evaluate it by applying it to all its arguments, the evaluation would take infinite time and memory. But

$$\text{fact} \langle 0 \rightarrow 1 \mid \langle n: \text{nat}+1. n \times \text{fact} (n-1) \rangle \rangle 5$$

applies the function to one argument, 5, and the evaluation takes finite time and memory.

The expression

$$\text{tree} \langle [\text{nil}], [\text{tree}; \text{int}; \text{tree}] \rangle$$

is all binary trees with integer nodes.

Here is a named-data expressing the infinitely long text $\infty * \text{"\heartsuit"}$ that is all hearts.

$$\text{hearts} \langle \text{"\heartsuit"}; \text{hearts} \rangle$$

A named-data is equal to the data with all occurrences of the name replaced by the named-data. So

$$\text{hearts} \langle \text{"\heartsuit"}; \text{hearts} \rangle$$

$$= \text{"\heartsuit"}; \text{hearts} \langle \text{"\heartsuit"}; \text{hearts} \rangle$$

$$= \text{"\heartsuit"}; \text{"\heartsuit"}; \text{hearts} \langle \text{"\heartsuit"}; \text{hearts} \rangle$$

and so on. Evaluation of $\text{hearts} \langle \text{"\heartsuit"}; \text{hearts} \rangle _ 5$ gives "\heartsuit" .

value-Data

A **value**-data allows us to use a program to compute data. It has the form

$$\text{value} \text{ simplename} : \text{data} := \text{data} \llbracket \text{program} \rrbracket$$

A local variable, called the **value**-variable, is defined with a type and initial value. Then the program is executed. The result is the final value of the **value**-variable. We have not yet presented programs, but the following example, which approximates the base of the natural logarithms e , should give the idea.

$$\text{value } \text{sum}: \text{rat} := 1$$

$$\llbracket \text{new } \text{term}: \text{rat} := 1.$$

$$\text{for } i: 1;..15 \llbracket \text{term} := \text{term}/i. \text{sum} := \text{sum} + \text{term} \rrbracket \rrbracket$$

There are no side effects. Nonlocal variables become constants within the program; their values may be used, but assigning them is not permitted. Input from and output to nonlocal channels are not permitted.

All the ways of expressing data can be combined arbitrarily, without restriction. Here we define *howeven* as a function whose body is a **value**-data. It expresses the number of times 2 is a factor of its argument.

$$\text{new } \text{howeven} \langle \langle n: \text{nat}+1. \text{value } h: 0;..n := 0$$

$$\llbracket \text{new } m: 1;..n+1 := n.$$

$$\text{loop } \llbracket \text{if } \text{even } m \llbracket h := h+1. m := m/2. \text{loop} \rrbracket \rrbracket \rrbracket \rangle$$

A **value**-variable begins its scope at the left program bracket `[[` and ends its scope at the corresponding right program bracket `]]` (see [Scope](#)). Consequently, the **value**-variable can be any simple name, even one that has already been defined in the scope that encloses the **value**-data. The type and initial value of the **value**-variable cannot use the **value**-variable.

Quote and Unquote

The predefined function *quote* takes any data and produces a text representation of the data. Roughly speaking, it quotes its argument. For examples, *quote* 123 = "123", *quote* \top = " \top ", *quote* "abc" = ""_abc"", *quote* {0, [1; 2]} = "{0,[1;2]}" . The argument is evaluated before quoting; for examples, *quote* (2 \times 3) = "6", *quote* (0<1) = " \top ". In a context where $x=3$, *quote* x = "3" and *quote* ($x\times 4$) = "12". Function *quote* does not provide any control over the format of the resulting text. For real numbers, fine control over the format of the resulting text is provided by the predefined function *realtex*.

The predefined function *unquote* is a kind of inverse of *quote*. It applies to texts; roughly speaking, it unquotes its argument. For examples, *unquote* "123" = 123, *unquote* " \top " = \top , *unquote* ""_abc"" = "abc", *unquote* "{0, [1; 2]}" = {0, [1; 2]}. The argument is evaluated after unquoting; for examples, *unquote* "2 \times 3" = 6, *unquote* "0<1" = \top . In a context where $x=3$, *unquote* " x " = 3 and *unquote* " $x\times 4$ " = 12. In a context where x is not defined, *unquote* " x " is not evaluated. If the argument does not represent a ProTem data expression, the function is not evaluated.

Whenever a data that is not a text is used in a context that requires a text, the *quote* function is applied automatically. For example

`! 123`

places a number where a text should be. So *quote* is applied automatically (`! quote 123`) resulting in a text (`!"123"`) as required for output to the screen. If *quote* has been redefined (see [Scope](#)), the predefined *quote* is used.

Whenever a text data is used in a context that requires a data that is not a text, the *unquote* function is applied automatically. For example,

`"123" + 1 = unquote "123" + 1 = 123+1 = 124`

If *unquote* has been redefined (see [Scope](#)), the predefined *unquote* is used.

Scope

A name is defined in these seven ways: by the keyword **new**, as a named-program, as a named-data, as a function parameter, as a plan parameter, as a **for**-index, or as a **value**-variable. We have already met function parameters and named-data and **value**-variables; we shall meet the others shortly. The scope of a name is the part of a program in which the name is defined. The scope of a name is limited by program brackets `[[]]`, except that the scope of a function's parameter is limited by function brackets `< >`, and the scope of the dataname of a named-data is limited by data brackets `()`.

A name defined in a local scope using the keyword **new** must be new, not already defined since the most recent opening program bracket `[[`. Its scope extends from its definition through all following sequentially composed programs to the corresponding closing program bracket `]]`. But it may be covered by a redefinition in an inner scope. Using `new x:= 2` and `new x:= 3` as example definitions, and letting *A*, *B*, *C*, *D*, and *E* stand for arbitrary program forms (but not **new** or

old), in

[[A. **new** x:= 2. B. [[C. **new** x:= 3. D]]. E]]

the definition of x as the number 2 is not yet in effect in A , but it is in effect in B , C , and E . The definition that makes x the number 3 is in effect in D . None of A , B , C , D , or E can contain a redefinition of x unless it is within further scope limiters `[[]]` or `< >` or `()`.

A name defined by **new** can become undefined by the keyword **old**, ending its scope early. So in

new x:= 2. A. **old** x. B

the definition of x is in effect in A but not in B . Within B , the name x has the same meaning (if any) that it had before the definition **new** x:= 2. After **old** x, the name x is again new and available for definition. However,

new x:= 2. [[**old** x. A]]

is not allowed; a scope cannot be ended by **old** within a subscope.

A scope can be nested inside another scope, which can be nested inside another, and so on. Outside all scope limiters `[[]]` and `< >` and `()` is the persistent scope. A name defined by **new** in the persistent scope is called a persistent name. Persistent names name data (documents) and programs that you want to keep. When you define a **new** name in the persistent scope, the name must differ from all names already in the persistent scope. The scope of a persistent name ends only with **old**. Its scope does not end with the end of a computing session (see [Session](#)), not even by switching off the power, which should not cut the power instantly, but should first cause the values of any variables in the persistent scope to be saved in nonvolatile memory. In the persistent scope, there is a dictionary named *predefined* that contains the predefined names.

Names must be defined before they are used (except for a [Forward Definition](#)). Whenever a name is used, it is looked up as follows: first, look in the most local scope (innermost `[[]]` or `< >` or `()`); then look in the next most local scope; then the next, and so on; when all local scopes have been searched, look in the persistent scope; finally, look in the *predefined* dictionary. The first occurrence found is the one that is used.

Programs

Some programs are concerned with names: defining a name (**new**), undefining a name (**old**). Other programs are variable assignment, input, output, and a variety of ways of combining programs to form larger programs. All programs, including those that define and undefine names, are executed in their turn, just like variable assignments and input and output.

Variable Definition

Variable definition has the form

new newname : data := data

The newname becomes a variablename. Here is an example variable definition.

new x: nat:= 5

This defines x to be a variable assignable to any element in *nat*, and initially assigned to 5. There is no such thing as an “uninitialized variable” nor the “undefined value” in ProTem. In a variable definition, the data after `:` is called the “type” of the variable, and the data after `:=` is called the “initial value”. The type can be anything except the empty bunch, and the initial value must be an element of the type. The type and initial value can depend on previously defined names, including variables. For example,

new y: 0,..2×x:= x

defines y as a variable whose value can be any natural number from (including) 0 up to (excluding) twice the current value of x (the value of x at the time this definition is executed), with initial value equal to the current value of x . But the type and initial value cannot make use of the variable being defined. ProTem allows the type and initial value to be “static” (known at compile time), as in the first example (variable x), and it allows the type and initial value to be “dynamic” (not known until execution time), as in the second example (variable y).

Here are five more examples.

```

new  $s$ :  $10*int:= 10*0$ 
new  $t$ :  $text:= ""$ 
new  $u$ :  $(0,..20)*char:= "abcde"$ 
new  $L$ :  $[*nat]:= [0;..10]$ 
new  $A$ :  $[3*[3*bin]]:= [3*[3*\top]]$ 

```

In the first example, s is defined as a variable that can be assigned to any string of ten integers, and is initially assigned to the string of ten zeroes. In the second example, $text$ is a predefined bunch equal to $*char$, so t can be assigned to any text, and is initially assigned to the empty text. In the next example, u is defined as a variable that can be assigned to any text of length less than 20, and is initially assigned to the text “abcde”. In the next example, L is defined as a variable that can be assigned to any list of naturals, and is initially assigned to the first ten naturals. In the last example, A is defined as a variable that can be assigned to a 3×3 array of binary values, and is initially assigned to a 3×3 array of \top .

Assignment

A variable can be reassigned by the assignment program. It has the form

```

variablename := data

```

Here are five examples using the definitions of the previous subsection.

```

 $x:= x+1$            \now  $x = 6$ 
 $s:= s<2>8$          \now  $s = 0; 0; 8; 0; 0; 0; 0; 0; 0; 0$ 
 $u:= u_(0,..3)$      \now  $u = "abc"$ 
 $L:= 3\rightarrow 5 \mid 5\rightarrow 7 \mid L$  \now  $L = [0; 0; 0; 5; 0; 7; 0; 0; 0; 0]$ 
 $A:= (1; 2)\rightarrow \perp \mid A$        \now  $A = [[\top; \top; \top]; [\top; \top; \perp]; [\top; \top; \top]]$ 

```

The data on the right of $:=$ must be an element in the type (evaluated at definition) of the variable on the left of $:=$. As in the examples, the data on the right of $:=$ can make use of the variable on the left of $:=$.

Constant Definition

Constant definition has the form

```

new newname := data

```

The newname becomes a constantname. The data on the right of $:=$ cannot make use of the name on the left of $:=$. The constantname cannot be reassigned. Here are three constant definitions.

```

new  $size:= 10.$ 
new  $piBy2:= pi / 2.$ 
new  $range:= 0,..size$ 

```

where pi is a predefined constant name.

A constant may use variables to express its value. For example

```

new  $xplus1:= x+1$ 

```

The current value of variable x is used to evaluate $x+1$, and $xplus1$ expresses that value. Variable

x may later be reassigned to another value, but that does not affect the value of $xplus1$. For example,

```

new  $x$ :  $nat := 3$ .      `  $x$  has value 3
new  $xplus1 := x+1$ .    `  $x$  has value 3 and  $xplus1$  has value 4
 $x := 5$                 `  $x$  has value 5 and  $xplus1$  has value 4

```

Data Definition

Data definition has the form

```
new newname ( data )
```

The newname becomes a dataname. Data definition gives the data a name, but does not evaluate the data.

```
new  $xplus2$  ( $x+2$ )
```

makes the value of $xplus2$ depend on the value of variable x . As x changes value, $xplus2$ changes value so that $xplus2 = x+2$ is always \top . In the constant definition of $xplus1$ earlier, $x+1$ is evaluated once, at definition time. In the data definition of $xplus2$, $x+2$ is not evaluated at definition time; it is evaluated every time $xplus2$ is used in a context that requires its value.

```

new  $x$ :  $nat := 3$ .      ` assigns  $x$  to 3
new  $xplus2$  ( $x+2$ ).    ` does not evaluate  $x+2$ 
 $x := xplus2$ .          ` assigns  $x$  to 5
 $x := xplus2$            ` assigns  $x$  to 7

```

A data definition can depend indirectly on a variable. For example,

```
new  $twoxplus4$  ( $2 \times xplus2$ )
```

makes $twoxplus4$ depend indirectly on the value of variable x . In this definition, the value of $xplus2$ is not required, so it is not evaluated. If x currently has value 3, then $x := twoxplus4$ assigns x to 10. Then another $x := twoxplus4$ assigns x to 24.

Data Recursion

In a variable definition, the type and initial value cannot depend on the variable being defined.

```
new  $bad$ :  $0, .. 2 \times bad := bad$   ` illegal
```

is not allowed due to the two occurrences of bad to the right of the colon. Likewise a constant definition cannot be recursive.

We have already seen that named-data can be recursive. Data definition also allows recursion. The next example defines div to be the integer division function for natural numbers.

```
new  $div$  ( $\langle a: nat. \langle d: nat+1. a < d \models 0 \Rightarrow even\ a \models 2 \times div\ (a/2)\ d \models 1 + div\ (a-d)\ d \rangle \rangle$ )
```

Here is a function that eats arguments until it is fed argument 0.

```
new  $eat$  ( $\langle n: nat. n=0 \models 0 \models eat \rangle$ )
```

So $eat\ 5\ 2\ 0 = 0$, and $eat\ 4\ 7\ 3\ 8\ 0 = 0$, and $eat\ 1\ 2 = eat$ and $eat: nat \rightarrow (0, eat)$.

Here is a baseless recursion. If it is evaluated, its evaluation is nonterminating.

```
new  $rec$  ( $rec$ )
```

Named-Data versus Data Definition

Since we have data definition, named-data is unnecessary. For example,

```
new  $t$ :  $tree$  ( $[[text], [tree; tree]] := ["abc"]$ )
```


concurrent composition remains a variable in at most one component of the concurrent composition; in all the other components of the concurrent composition, it becomes a constant.

new $a: nat := 1 \parallel$ **new** $b: nat := 2$. **new** $c (a+b)$. $[[a := 4. A]] \parallel [[b := 8. B]]. C$

In the concurrent composition $[[a := 4. A]] \parallel [[b := 8. B]]$, variable a can be reassigned in one of the concurrent programs, but not in both. Likewise variable b can be reassigned in one of the concurrent programs, but not in both. At the start of A , variable a has value 4, constant b has value 2, and data c has value 6. At the start of B , constant a has value 1, variable b has value 8, and data c has value 9. If A does not reassign a , and B does not reassign b , then at the start of C , variable a has value 4, variable b has value 8, and data c has value 12.

new $a: nat := 1 \parallel a := 2$ `illegal

is not allowed because the use of a does not sequentially follow its definition. Concurrent programs cannot affect each other through assignments of variables. For co-operation, programs can communicate with each other on channels (see [Channel Definition](#)). A channel can be used for output in only one component of a concurrent composition. It can be used for input in all components, reading the same inputs independently.

Program brackets $[[\]]$ are needed for a concurrent composition of sequential compositions

$[[A. B]] \parallel [[C. D]]$

because concurrent composition comes before sequential composition in the execution order.

In summary: A name defined in one part of a concurrent composition cannot be used in the other parts, but it can be used in a sequentially following program. A variable defined before a concurrent composition remains a variable in at most one part of the composition, and is a constant in the other parts. A channel defined before a concurrent composition can be used for output in at most one part of the composition, and can be used for input in all parts.

[if-Program](#)

An **if**-program has the form

if data $[[$ program $]]$

The **if**-program

if b $[[P]]$

is executed as follows: binary expression b is evaluated; if its value is \top , then program $[[P]]$ is executed; if its value is \perp , then program $[[P]]$ is not executed. An **if-else**-program has the form

if data $[[$ program $]]$ **else** $[[$ program $]]$

The **if-else**-program

if b $[[P]]$ **else** $[[Q]]$

is executed as follows: binary expression b is evaluated; if its value is \top , then program $[[P]]$ is executed and program $[[Q]]$ is not executed; if its value is \perp , then program $[[P]]$ is not executed and program $[[Q]]$ is executed. The program **if** b $[[P]]$ is equivalent to **if** b $[[P]]$ **else** $[[ok]]$ where ok is a predefined program whose execution does nothing and takes no time.

[case-Program](#)

A **case**-program has the form

case data $[[$ program $]]$

in which the program is a sequence of cases. For example, in a **case**-program with three cases,

case n $[[P]]$ $[[Q]]$ $[[R]]$

the programs, called cases, are numbered in order. Case 0 is $[[P]]$, case 1 is $[[Q]]$, and case 2 is $[[R]]$. It is executed as follows: natural expression n is evaluated; then one of the programs is

executed. In the example, if n has value 1, then just $[[Q]]$ is executed. If n is equal to or greater than the number of cases, an error message is printed on channel msg and execution stops. The example **case**-program is equivalent to

```
if n=0 [[P]] else [[if n=1 [[Q]] else [[if n=2 [[R]] else [!"Error: case index too large.". stop]]]]
```

Here is another example, also with three cases:

```
case 2-n
[[ok]] `case 0: all is well, nothing to do
[!"warning: don't do it again"] `case 1: warning
[!"you're fired"] `case 2: firing
```

As the example illustrates, it is helpful to put a comment with each case to say the case index.

A **case-else**-program has the form

```
case data [[ program ]] else [[ program ]]
```

in which the first program is a sequence of cases. A **case-else**-program with three cases,

```
case n [[P]] [[Q]] [[R]] else [[S]]
```

is the same as the **case**-program, but if $n \geq 3$, then the program $[[S]]$ after **else** is executed. The example **case-else**-program is equivalent to

```
if n=0 [[P]] else [[if n=1 [[Q]] else [[if n=2 [[R]] else [[S]]]]]
```

for-Program

A **for**-program has the form

```
for simplename : data [[ program ]]
```

The simplename becomes a constantname in the program. Here is a nest of **for**-programs that computes the transitive closure of array A : $[n*[n*bin]]$.

```
for j: 0;..n [[for i: 0;..n [[for k: 0;..n [[if A i j  $\wedge$  A j k [[A:= (i;k)  $\rightarrow$   $\top$  | A]]]]]]]
```

The **if**-program $\text{if } A i j \wedge A j k \text{ } [A:= (i;k) \rightarrow \top | A]$ can be restated as

$$A:= (i;k) \rightarrow (A i k \vee (A i j \wedge A j k)) | A$$

if you prefer. The name being defined by **for** is called a **for**-index. It is known only within the **for**-program, and it is known there as a constant, and so it is not assignable. In the example, each of the **for**-indexes j , i , and k takes values 0, 1, 2, and so on up to but excluding n .

For a second example, here is the sieve of Eratosthenes.

```
new n:= 1000. new prime: n*bin:= 2* $\perp$ ; (n-2)* $\top$ .
for i: 2;..ceil (sqrt n) [[if prime_i [[for j: i;..ceil (n/i) [[prime:= prime <i>xj>  $\perp$ ]]]]]
```

A **for**-index is “by initial value”, so this example increases x by 1, not 2.

```
for i: x; x [[x:= i+1]]
```

This next example prints the natural numbers forever.

```
for n: 0;.. $\infty$  [!" n; “ ”]
```

After the `:` we can have any string expression; the **for**-index stands for each item in the string, in sequence. We can also have any bunch expression; the **for**-index stands for each element of the bunch, concurrently. As an example (note the use of `,..` rather than `;` as earlier),

```
for i: 0;..#L [[L:= i  $\rightarrow$  0 | L]]
```

makes the items of L be 0, concurrently. We could also write either of these:

```
for i:  $\square$ L [[L:= i  $\rightarrow$  0 | L]]
```

```
L:= [#L*0]
```

The domain of the **for**-index can also be a bunch of strings, or a string of bunches, and so on, so that sequential and concurrent execution can be nested within each other. (Note: distribution and factoring laws are not applied; the structure of the expression is the structure of execution.)

A **for**-index begins its scope after `[[` and ends its scope at the corresponding `]]`. Consequently, the **for**-index can be any simple name, even one that has already been defined in the scope that encloses the **for**-program. The domain of the **for**-index cannot use the **for**-index.

Program Definition

Program definition has the form

new newname `[[program]]`

The newname becomes a programname within the program. Program definition gives a program a name, but does not execute the program. For example,

new *switchends* `[[L:= 0 → L 9 | 9 → L 0 | L]]`

Execution of this definition defines the program name *switchends*, but does not execute program `[[L:= 0 → L 9 | 9 → L 0 | L]]`. After execution of this definition, the name *switchends* can be used to call (cause execution of) the program it names. Program definitions can be recursive. Predefined program names include *await*, *exec*, *ok*, *stop*, *wait*.

A fancy name can be used as a specification. For example,

new `« x' > x »` `[[x:= x+1]]`

The specification `« x' > x »` means the final value of *x* is greater than its initial value. It is implemented (refined, implied) by the program `[[x:= x+1]]`. A prover is invoked by the `ctl v` command (see [Verify](#)). If the specification is written within the language that the prover understands, the prover attempts to prove that the specification is implemented (refined, implied) by the program. If the program makes use of a specification, the inner specification is used in the outer proof. For example,

new `« x' = 0 »` `[[if x≠0 [[x:= x-1. « x' = 0 »]]]]`

In the program `[[if x≠0 [[x:= x-1. « x' = 0 »]]]]`, the specification `« x' = 0 »` means exactly what it says, rather than the program that it names. Thus the use of specifications makes complicated fixed-point semantics unnecessary. If the prover fails to understand the specification, or fails to prove the refinement, it informs the programmer, and treats the specification as just a name. (See the paper [Specified Blocks](#).)

Named-Program

A named-program has the form

simplename `[[program]]`

The simplename becomes a programname within the program that it names. Although the name is situated before a `[[`, it begins its scope after `[[` and ends its scope at the closing `]]`. Consequently, the name can be any simple name, even one that has already been defined in the scope that encloses the named-program. The name is attached to the program (like a program definition), and the program is executed (unlike a program definition). One purpose of this naming is to make loops. Here is a two-dimensional search for *x* in an $n \times m$ array *A* of integers (that is, $A: [n*[m*int]]$).

```

new i: nat:= 0.
tryThisI [if i=n [! x; “ does not occur.”]]
  else [new j: nat:= 0.
    tryThisJ [if j=m [i:= i+1. tryThisI]]
      else [if A i j = x [! x; “ occurs at ”; i; “ ”; j]
        else [j:= j+1. tryThisJ]]]]]]

```

The next example is a fast remainder program, assigning natural variable r to the remainder when natural a is divided by positive natural d , using only addition and subtraction.

```

r:= a.
outerloop [if r≥d [new dd: nat:= d.
  innerloop [r:= r-dd. dd:= dd+dd.
    if r<dd [outerloop] else [innerloop]]]]]]

```

$N[P]$ is equivalent to $[P]$ with occurrences of N replaced by $N[P]$. So the fast remainder example means the same as

```

r:= a.
[if r≥d [new dd: nat:= d.
  [r:= r-dd. dd:= dd+dd.
    if r<dd [outerloop [if r≥d [new dd: nat:= d.
      innerloop [r:= r-dd. dd:= dd+dd.
        if r<dd [outerloop] else [innerloop]]]]]]]]
    else [innerloop [r:= r-dd. dd:= dd+dd.
      if r<dd [outerloop [if r≥d [new dd: nat:= d.
        innerloop [r:= r-dd. dd:= dd+dd.
          if r<dd [outerloop]
            else [innerloop]]]]]]]]
      else [innerloop]]]]]]]]

```

This process can be repeated. Although semantically there are calls, in the previous two examples they are last actions (tail recursions), so they are implemented as branches (jumps, go to's).

The next example illustrates that named programs provide general recursion, not just tail recursion. It computes the Fibonacci numbers $x:=fib\ n$ and $y:=fib\ (n+1)$ in $\log n$ time.

```

Fib [if n=0 [x:= 0. y:= 1]]
  else [if odd n [n:= (n-1)/2. Fib. n:= x. x:= x^2 + y^2. y:= 2×nxy + y^2]]
  else [n:= n/2 - 1. Fib. n:= x. x:= 2×xy + y^2. y:= n^2 + y^2 + x]]]]

```

As in a program definition, a fancy name can be used as a specification. For example,

```

« x' > x » [x:= x+1]

```

The specification « $x' > x$ » is implemented (refined, implied) by the program [$x:=x+1$]. A prover is invoked by the `[ctl v]` command (see [Verify](#)). If the specification is written within the language that the prover understands, the prover attempts to prove that the specification is implemented (refined, implied) by the program. If the program makes use of a specification, the inner specification is used in the outer proof. For example,

```

« x' = 0 » [if x≠0 [x:= x-1. « x' = 0 »]]

```

Inside the program brackets, the specification « $x' = 0$ » means exactly what it says, rather than the program that it names. Thus the use of specifications makes complicated fixed-point semantics unnecessary. If the prover fails to understand the specification, or fails to prove the refinement, it informs the programmer, and treats the specification as just a name. (See [Specified Blocks](#).)

Suppose a name is defined within a loop. For example, the name *a* in
infiniteLoop **[[new a:= "a". !a. infiniteLoop]]**

Executing this loop prints an infinite sequence of the letter "a". Replacing the call with the named-program, it is equivalent to

[[new a:= "a". !a. infiniteLoop [new a:= "a". !a. infiniteLoop]]]

In a general recursion, each call opens a new scope, and each new definition hides but does not destroy the previous definition. But when the recursive call is the last action performed in the named-program (a tail recursion), as in this example, the old scope and its definitions cannot be used again, so the new scope replaces the old one; the scopes and variables do not pile up.

Named-Program versus Program Definition

Let *name* be a new name (not defined in the local scope), and let *program* be a program, possibly using the name *name*. Then the following three lines are equivalent to each other.

name **[[program]]**

[[new name [[program]]. name]]

new name [[program]]. name. old name

Therefore named-programs are unnecessary. For example

loop **[[if n>0 [[n:= n-1. loop]]]]**

is equivalent to

[[new loop [[if n>0 [[n:= n-1. loop]]]]. loop]]

A named-program is a succinct way of recursively defining a program, and using the program once. However, if you need to use (call) a program more than once, you must make a program definition.

Output and Input

Each channel is defined to transmit a specific type of value. The output channels *screen*, *printer*, and *mail*, and the input channels *keys* and *mail* are predefined to transmit text. The input channel *microphone* and the output channel *speaker* are predefined to transmit sound. We can define local channels to transmit any type of value (see [Channel Definition](#)).

Output has the form

channelname ! data

Channel *screen* accepts text, which is displayed on the screen. The program

screen! "Hi there."

sends the text "Hi there." to the screen. Output is buffered so it will be available when *screen* is ready to receive it. Texts can be joined and sent together.

screen! "Answer = "; *quote* x; *nl*

where *quote* is a predefined function that converts to a text, and *nl* is the new line character, or next line character, or return character. Function *quote* can be omitted (see [Quote and Unquote](#)).

When the *delete* (backspace) character is output to the *screen* or *printer*, the previous character is deleted. If there are more *delete* characters than previous characters, the extra *delete* characters are ignored. When the *tab* character is output, some amount of space is substituted. When the *nl* character is output, further characters start on a new line.

Email input and output are made convenient for nonprogrammers by a mail program, but at the level of ProTem programming, emailing looks like this:

```
mail! "To: hehner@cs.utoronto.ca"; nl;
      "Hey Rick, have a look at the exec program: "; nl;
      definition "exec"
```

The keyboard is a program that runs concurrently with other programs; you do not need to initiate it; it is already running. It monitors what key combinations are pressed, and for what duration, and outputs a string of characters (a text) on channel *keys*. The `[shift A]` combination is a single character "A". The escape key and the `[ctl.]` combination are both the single character named *end*. The click button is just a key like any other; *click* and *doubleclick* are characters.

Input has two forms: without echo, and with echo. The form without echo is

```
channelname ? data ( ; data ) data
```

The channelname is the input channel. The input channel may come from a keyboard or microphone or camera, or it may come from a document, or it may be an internal channel (see [Channel Definition](#)). The type of input read is determined by the channel. The input read is the earliest input on the channel that has not yet been read. If input is not yet available, it is awaited. What comes after `?` is called the pattern. The pattern directs the input. The pattern is in 3 parts. The first part is called the left context; the middle part between the core brackets `(;)` is the part we want, called the core; and the last part is called the right context. After the input program, we refer to the input matching the core using the channel name followed by `?`. If the available input is shorter than required by the pattern, execution awaits further input. If the available input is longer than required by the pattern, the remaining input is future input.

Here is an example. The constant *digit* is predefined as

```
new digit:= "0", "1", "2", "3", "4", "5", "6", "7", "8", "9"
```

Let *doc* be an input channel that reads a text document. Then

```
doc? *" " (digit) ""
```

reads input from channel *doc*. First, there may be some spaces `*" "`, and we want to pass over them. Then there is a *digit*, and that's what we want. After that, we don't want to read any more input, so that's the empty string `""`. The digit we read can be referred to as *doc?*. For example,

```
if doc?="5" [[screen! "You win."]]
```

Here is a further input from channel *doc*.

```
doc? "" (text) nl
```

This input reads whatever remains of the current line, up to and including *nl* (the new line character). Then *doc?* refers to the text up to but not including *nl*.

The space, *tab*, *nl*, *delete*, and *end* characters may be part of the input; they are just characters like any others. For example, to read one character and then output a visible character,

```
keys? "" (char) "". if keys?=" " [[screen! "␣"]]
                    else [[if keys?=tab [[screen! "␣"]]
                          else [[if keys?=nl [[screen! "↵"]]
                                else [[if keys?=delete [[screen! "␣"]]
                                      else [[if keys?=end [[screen! "■"]]
                                            else [[screen! keys?]]]]]]]]]]
```

After the input

```
keys? *(" ", tab, nl) (*digit) end
```

the digits can be referred to as *keys?*. Then we might have the assignment

```
x:= unquote (keys?) + unquote (keys?)
```

where *unquote* is a predefined function that converts from a text to the type required. Function

unquote can be omitted (see [Quote and Unquote](#)), so we may write

```
x:= keys? + keys?
```

Both occurrences of *keys?* refer to the same input, and *keys?* continues to refer to the same input until the next time new input is read from channel *keys* .

The left context is as long as possible. In

```
keys? *(" ", tab, nl) (text) end
```

the left context is **(" ", tab, nl)* . This consumes spaces, tabs, and new line characters, until the first character that is none of those. The core is as short as possible while allowing the right context, which is *end* , to match. So the core consists of all characters up to but not including the first occurrence of *end* . The right context is also as short as possible. So

```
keys? "" (text) ""
```

just inputs the empty text no matter what input is available.

Pattern *numpat* is predefined as follows.

```
new numpat:= ("+", "-", ""); digit; *digit; (("."; digit; *digit), "")
```

To receive a text on channel *doc* that can be interpreted as a number, possibly preceded by spaces, ending in a space or comma or new line or *end* , input

```
doc? *"" (numpat) (" ", ",", nl, end)
```

If the right context *(" ", ",", nl, end)* were *""* , only leading spaces and an optional sign and the first digit would be read; subsequent digits, a point, and more digits would not be read.

If *c* is a channel to input text, the program

```
c? "" ( "y", "n" ) ""
```

inputs one character, either "y" or "n" , from channel *c* . If the first available character on channel *c* is "a" , or more generally, if the input on the channel does not fit the pattern, what happens is not defined. Here are three options.

- The program cannot be executed, so execution ends.
- An error message is sent to channel *msg* to say that the input is unacceptable, and execution ends.
- An error message is sent to channel *msg* to say that the input is unacceptable, and the sender is given another opportunity to send an input that fits the pattern.

What happens depends on the implementation; it may depend on the channel, and on the number of attempts. Perhaps the last option is appropriate for channel *keys* , and the first is appropriate for a secure channel. Of course, you can read whatever character you are given *c? "" (char) ""* and decide what happens if it is neither "y" nor "n" .

Input without echo is invisible. It is useful for reading from a stored document. It is useful when reading a password (see [Read Password](#)). It is useful for single-stepping through an execution. But it is not useful for ordinary keyboard input. It does not allow corrections by deletions and editing; instead, *delete* and *click* are just characters like any others. Input that allows corrections requires a visible echo, which we introduce next.

Input with echo has the form

```
channelname ? data ( data ) data ! channelname
```

What follows *!* is the output channel for the echo. Each input item is immediately echoed (output) on the echo channel. It allows you to use *delete* and cut/copy/paste to change what you have entered. The echo shows each character as it is entered, and each deletion and edit as it is made. When the pattern has been fulfilled, the input is finished. For example,

```
keys? "" (text) end !screen
```

reads text until *end* is sent. The input may span several lines, and contain *tab* and *nl* (new line) characters. Characters entered can be deleted (including new line characters), and changes can be made using cut/copy/paste, until *end* is sent, fulfilling the pattern. Then the corrected text, not including *end*, can be referred to as *keys?*.

If *c* is the name of an input channel, then *c??* is a binary expression with value \top if there is written but unread data on the channel, and \perp if there is not. For example,

```
if keys?? [[keys? "" (char) "" !screen]] else [[screen! "Are you still there?"]]
```

If *doc* is a channel that reads from a document, *doc??* tells whether there is more to be read. Input from a document channel that does not currently have any written-but-not-yet-read data waits until data is written to the channel by a concurrent program.

If *c* is a channel to input text and *d* is a channel to output text, the program

```
c? "" ( "y", "n" ) "" !d
```

inputs one character, either "y" or "n", from channel *c*, and echoes it to channel *d*. If the first available character on channel *c* is "a", or more generally, if the input on the channel does not fit the pattern, execution waits for a correction to make the input fit the pattern. Input with echo allows correction; input without echo does not.

Suppose the input program *keys? "" (text) "z" !screen* is executed. And suppose the input "ab" is keyed in. And then the input is edited by inserting "z" between "a" and "b". The left context is empty, the core is "a", the right context is "z", and "b" is future input. (See predefined *drain*.)

The two programs

```
in? left (core) right !out
```

```
in? left (core) right. out! in?
```

usually differ. The first echoes all the input matching *left*, *core*, and *right*, and allows input deletion and editing until the pattern is fulfilled. The second treats *delete* and *click* as ordinary characters being read, without deletions or editing, and outputs only the input matching *core*.

We have now seen all of input and output. But there are some abbreviations to make them more convenient. An output channel name can be omitted, in which case the output channel is *screen*. For example,

```
! "Hello World"
```

prints Hello World on channel *screen*. And

```
! 2+2
```

prints 4 on channel *screen*. This program is short for

```
screen! quote (2+2)
```

If the output channel is omitted, and the name *screen* has been redefined, the output channel is the predefined channel *screen*.

An input channel name can be omitted, in which case the input channel is *keys*. For example,

```
? "" (text) end!
```

reads text from *keys* (possibly corrected, terminated by *end*), with echo to *screen*. The most recent core input on channel *keys* can be referred to as just *?*. And *??* is a binary expression saying whether there is written but unread data on channel *keys*. If the input channel is omitted, and the name *keys* has been redefined, the input channel is the predefined channel *keys*.

The expression *unquote (keys?)* can be written *unquote (?)* or *keys?* or *?*. But it cannot be written *unquote keys?* because that is parsed as *(unquote keys)?*. And it cannot be written

unquote ? because the implementation will complain that *unquote* is not a channel. Similarly for *quote* (*keys*??) . When *keys*? is shortened to ? and used as a list index or as an argument to a function or plan, it must be in evaluation brackets (?) . When *keys*?? is shortened to ?? and used as an argument to a function or plan, it must be in evaluation brackets (??) .

In input with echo, the pattern can be omitted, in which case the pattern is

```
*(" ", tab, nl) (text) end
```

The shortest input program

```
?!
```

is short for

```
keys? *(" ", tab, nl) (text) end !screen
```

and means: from channel *keys* , skip leading spaces, tabs, and new lines, then read text including spaces, tabs, and new lines, corrected by any deletions and editing, until the *end* character is received. The value of *keys*? , or just ? , is the core text, not including the leading spaces, tabs, and new lines, and not including the trailing *end* . The entire input is echoed, each character, each deletion, each editing change, on channel *screen* .

The shortest output program

```
!?
```

is short for

```
screen! keys?
```

and means: on channel *screen* , output the core of the most recent input from channel *keys* .

In summary, output is

```
outchannel ! data
```

Input without echo

```
inchannel ? leftcontext ( core ) rightcontext
```

reads data according to the pattern, including such characters as *delete* and *click* , with no corrections and no echo. Input with echo

```
inchannel ? leftcontext ( core ) rightcontext ! echochannel
```

is correctable by deletion and editing until the pattern is fulfilled. The core input most recently read on inchannel is referred to as

```
inchannel ?
```

The binary expression

```
inchannel ??
```

says whether there is written but unread data on inchannel. If the channel name is *keys* or *screen* it can be omitted. In an input with echo, if the pattern is

```
*(" ", tab, nl) (text) end
```

it can be omitted.

Channel Definition

Channel definition has the form

```
new newname ? data ! data
```

The newname becomes a channelname. The first data is the type, and the second data is the initial value. The type and initial value cannot use the name of the channel being defined. The definition

```
new c? nat !nil
```

defines *c* to be a new local channel that transmits values of type *nat* , and initially there are no values in the channel. Initially *c?=nil* because no input has yet been read. Initially *c??=⊥* to say there is no written but unread data in the channel. Channel *c* can be used for output and input.

The definition

```
new updown? text !"What goes up "
```

creates a channel named *updown* that transmits text. Initially $updown?=""$ because no input has yet been read. Initially $updown??=\top$ because there is the written but unread input

```
"What goes up "
```

After the input program

```
updown? "" (char) ""
```

we have $updown?="W"$. Then, after the output program

```
updown! "must come down."
```

the unread input available for reading on channel *updown* is

```
"hat goes up must come down."
```

Now, after the input program with echo

```
updown? "" (char) "" !updown
```

we have $updown?="h"$, and the unread input available on channel *updown* is

```
"at goes up must come down.h"
```

If we have a text document named *report*, we can create a channel *rpt* to read from it like this:

```
new rpt? text !report
```

Initially $rpt?=""$. If document *report* is nonempty, then initially $rpt??=\top$.

If channel *c* is defined to have type *T*, it holds a string of values of type *T*, which is a value of type $*T$. A string of values can be read together in one pattern, so $c?$ also has type $*T$. For example,

```
new nn? nat !5; 3; 4
```

defines *nn* as a channel that transmits natural numbers. Then $nn? nil (2*nat) nil$ reads 2 items from the channel, after which $nn?=5; 3$. If the definition had been

```
new nn?  $*nat$  !5; 3; 4
```

the result is the same, because $**T = *T$. Since $text = *char$, channels *updown* and *rpt* could equally well have type *char*.

At the beginning of a session (see [Session](#)), all persistent channels are redefined (reinitialized). Before any keyboard input has been keyed in, $keys?=""$ and $keys??=\perp$.

To use a channel for communication between concurrent programs, define the channel before the start of the concurrent programs. Then one of the concurrent programs can write to the channel, and all the concurrent programs can read from the channel; they read the same inputs independently. If a program sequentially follows a concurrent composition, the sequentially following program begins reading on each channel after all the inputs read by all concurrent programs on that channel. For example,

```
new c? nat ! nil. AllB. D
```

On channel *c*, program *D* begins reading after the last input read by *A* or *B*.

```
new c? nat ! nil. x:=c?
```

```
\assigns x to nil
```

```
new c? nat ! nil. c? nil (nat) nil
```

```
\infinite wait for input
```

```
new c? nat ! nil. c! 7. c? nil (nat) nil. x:=c?
```

```
\assigns x to 7
```

```
new c? nat ! nil. [c? nil (nat) nil. x:=c?] || c! 7
```

```
\assigns x to 7
```

```
new c? nat ! nil. c! 7. c! 8. c? nil (nat) nil. x:=c?
```

```
\assigns x to 7
```

```
new c? nat ! nil. c! 7. [c? nil (nat) nil. x:=c?] || y:=c?
```

```
\assigns x to 7 and y to nil
```

```
new c? nat ! nil. c! 7. [c? nil (nat) nil. x:=c?] || [c? nat. y:=c?]
```

```
\assigns x and y to 7
```

Plan

A plan is a program with a parameter. There are five forms of plan. The first is

plan simplename : data [[program]]

The simplename is being defined as a constant parameter within the program. It can be any simple name, even one that has already been defined in the current scope. Its type (after the :) cannot make use of the parameter. The scope of the parameter is from [[to]]. For example,

plan y: real [[x:= xxy]]

A plan can be argued by adjacency in the same way that lists are indexed and functions are argued. The argument provides a value for the parameter. For example,

plan y: real [[x:= xxy]] 3

is the same as

x:= x×3

Commonly, a plan is named

new P [[**plan** y: real [[x:= xxy]]]]

and then argued P 3 , but a plan is not required to have a name.

A program with $n+1$ parameters is a program with 1 parameter whose body is a program with n parameters. For example, here is a program with two parameters.

plan x: int [[**plan** y: int [[z:= x+y]]]]

Each argument reduces the number of parameters.

plan x: int [[**plan** y: int [[z:= x+y]]]] 3 4

is equivalent to

plan y: int [[z:= 3+y]] 4

which is equivalent to

z:= 3+4

A plan can be named (in a program definition or named-program); a plan can be argued; and a plan can be the body of a plan. A plan that is not fully argued cannot be executed. A plan that has been fully argued can be used wherever any program can be used. For examples,

plan x: int [[**plan** y: int [[z:= x+y]]]]. z:= 2 `this is not allowed

plan x: int [[**plan** y: int [[z:= x+y]]]] 3. z:= 2 `this is not allowed

plan x: int [[**plan** y: int [[z:= x+y]]]] 3 4. z:= 2 `this is allowed

Here is a named-program to find the maximum value in nonempty list L in $\log(\#L)$ time. (L is a variable, and its initial value is destroyed in the process.) We define $findmax\ i\ j$ to find the maximum in the segment of L from (including) index i to (excluding) index j , reporting the result as $L\ i$, and then apply it to $0(\#L)$, which makes $L\ 0$ the maximum value of the initial list.

$findmax$ [[**plan** i: □L [[**plan** j: □L+1
 [[if $j-i \geq 2$ [[$findmax\ i\ (div\ (i+j)\ 2)$ || $findmax\ (div\ (i+j)\ 2)\ j$.
 $L := i \rightarrow L\ i \vee L\ (div\ (i+j)\ 2) \mid L]]]]]] 0(\#L)$

In the previous paragraphs, the parameter is a constant (note the :); it is not assignable. It is “by initial value”, so

plan i: int [[x:= i. y:= i]] (x+1)

assigns both x and y to the same value, one more than x 's initial value.

The second form of plan

plan simplename := data [[program]]

(note the :=) defines a variable parameter. For example,

plan $x := \text{int} \llbracket x := 3 \rrbracket$

A plan with a variable parameter applies to a variable argument. But it cannot be applied to a variable appearing in the plan. This restriction is required for reasoning about the plan. This example plan can be applied to any variable, even one named x , because that x (the argument) is nonlocal, and is not the local variable x (the parameter) appearing in the plan. But the plan

plan $x := \text{int} \llbracket x := 3 \parallel y := 4 \rrbracket$

cannot be applied to variable y .

Here is a plan named *norm* to reduce rational num/denom to lowest terms.

new norm $\llbracket \text{plan } \text{num} := \text{nat}+1 \llbracket \text{plan } \text{denom} := \text{nat}+1 \text{ ` normalize } \text{num}/\text{denom}$
 $\llbracket \text{new } g := \text{gcd} (\langle a: \text{nat}+1. \langle b: \text{nat}+1. \text{ ` greatest common divisor of } a \text{ and } b$
 $a=b \models a = a < b \models \text{gcd } a (b-a) \models \text{gcd } (a-b) b \rangle \rangle \text{num } \text{denom}.$
 $\text{num} := \text{num}/g. \text{denom} := \text{denom}/g \rrbracket \rrbracket$

If variables x and y have values 8 and 12, then after *norm* $x y$ they have values 2 and 3.

The main use for variable parameters is probably to affect many files in the same way; for example, a plan to sort the contents of files.

The next form of plan

plan *simplename* ! *data* $\llbracket \text{program} \rrbracket$

creates a plan with an output channel parameter. For example.

plan $c! \text{ text} \llbracket c! \text{ "abc"} \rrbracket$

A plan with a channel parameter cannot be applied to a channel appearing in the plan. This example plan can be applied to any output channel that receives text, even one named c , because that c (the argument) is nonlocal, and is not the local channel c (the parameter) appearing in the plan. But

plan $c! \text{ text} \llbracket c! \text{ "abc"} \parallel d! \text{ "def"} \rrbracket$

cannot be applied to channel d . The channel name *screen* cannot be omitted when used as an argument for an output channel parameter.

The next form of plan

plan *simplename* ? *data* $\llbracket \text{program} \rrbracket$

creates a plan with an input channel parameter. For example.

plan $c? \text{ text} \llbracket c?!d \rrbracket$

This plan applies to an input channel that delivers text, but not to a channel appearing in the plan.

plan $c? \text{ text} \llbracket c?!d \parallel d?!c \rrbracket$

cannot be applied to channel d . The channel name *keys* cannot be omitted when used as an argument for an input channel parameter.

The following program *pps* has three channel parameters. On the first, a , it reads the coefficients of a rational power series; on the second, b , it reads the coefficients of another rational power series; on the last, c , it writes the coefficients of the product power series.

new pps $\llbracket \text{plan } a? \text{ rat} \llbracket \text{plan } b? \text{ rat} \llbracket \text{plan } c! \text{ rat}$
 $\llbracket a? \text{ nil } (\text{rat}) \text{ nil} \parallel b? \text{ nil } (\text{rat}) \text{ nil}. c! a? \times b?$
 $\text{new } a0 := a? \parallel \text{new } b0 := b? \parallel \text{new } d? \text{ rat } !\text{nil}.$
 $\text{pps } a \ b \ d$
 $\parallel \llbracket a? \text{ rat} \parallel b? \text{ rat}. c! a0 \times b? + a? \times b0.$
 $\text{loop} \llbracket a? \text{ nil } (\text{rat}) \text{ nil} \parallel b? \text{ nil } (\text{rat}) \text{ nil} \parallel d? \text{ nil } (\text{rat}) \text{ nil}.$
 $c! a0 \times b? + d? + a? \times b0. \text{loop} \rrbracket \rrbracket \rrbracket$

The final form of plan

plan *simplename* \ [[*program*]]
 creates a plan with a dictionary parameter. This plan can be applied to any dictionaryname that does not appear in the plan.

Dictionary Definition

Dictionaries are the way you organize your programs and data. There are two forms of dictionary definition. The first form is

new *newname* \

The *newname* becomes a dictionaryname. To create a new dictionary named *abc*, write

new *abc* \

The newly created dictionary is empty. Now you can define names within this dictionary. For example,

new *abc* \ *x* := 2

defines *x* in dictionary *abc* to be the constant 2. This constant can then be used as *abc* \ *x*. (It does not matter whether there are spaces before or after a backslash.) A name being defined in a dictionary must not already be defined in that dictionary in the current scope. If the current scope is a local scope, the name becomes undefined at the end of the scope, as usual. Each name in a dictionary is defined, using the keyword **new** and a compound name, to be one of the following: a variable name, a constant name, a data name, a program name, a channel name, a unit name, or a dictionary name. To define new dictionary *def* within dictionary *abc* write

new *abc* \ *def* \

When a name in a dictionary is defined to be a dictionary, this dictionary also can contain names, some of which can be defined as dictionaries, and so on. So a dictionary can be a tree structure. Suppose there is a dictionary named *ProTem* within which there is a dictionary named *grammars* within which there is a text named *LL1*. Its name is *ProTem* \ *grammars* \ *LL1*.

A name within a dictionary can be undefined by **old** in the same scope where it was defined (see [Name Removal](#)). For example, **old** *abc* \ *x* ends the definition of *x* in dictionary *abc*. In the scope where *abc* was defined, **old** *abc* ends the definition of dictionary *abc*. When a name becomes undefined, what it named remains in existence, anonymously, as long as something refers to it. When a dictionary becomes undefined, so do all the names within it. Here is an example.

new *stack* \.

new *stack* \ *s* : **nat* := *nil*.

new *stack* \ *push* [[**plan** *x* : *nat* [[*stack* \ *s* := *stack* \ *s*; *x*]]].

new *stack* \ *pop* [[*stack* \ *s* := *stack* \ *s* (0; .. ↔ *stack* \ *s* - 1)].

new *stack* \ *top* ((*stack* \ *s* (↔ *stack* \ *s* - 1)).

old *stack* \

Dictionary *stack* now has three visible names in it: *push*, *pop*, and *top*. Variable *s* still exists, but it is hidden.

The second form of dictionary definition is

new *newname* \ \ *dictionaryname*

We can create a dictionary named *parseStack* populated with the same definitions as *stack*.

new *parseStack* \ \ *stack*

It is equivalent to writing

```

new parseStackλ.
  new parseStackλs: *nat:= nil.
  new parseStackλpush [[plan x: nat [[parseStackλs:= parseStackλs; x]]].
  new parseStackλpop [[parseStackλs:= parseStackλs_(0;..parseStackλs-1)].
  new parseStackλtop (parseStackλs_(↔parseStackλs-1)).
  old parseStackλs

```

There is a dictionary in the persistent scope named *predefined* (see [Predefined Names](#)).

[Measuring Unit Definition](#)

There are three predefined units of measurement. They are *g*, representing mass in grams, *m*, representing distance in meters, and *s*, representing time in seconds. A unit of measurement has all the properties of an unknown positive finite real number constant. So, for example, we write $10 \times m/s$ for the speed 10 meters per second. And we can define

```
new km:= 1000×m
```

to make *km* be a kilometer, and

```
new h:= 60×60×s
```

to make *h* be an hour. So $1 \times m/s = 3.6 \times km/h$ evaluates to \top . To assign a variable to a quantity with units attached, the variable's type must have compatible units attached. For example,

```
new speed: real×m/s:= 3.6×km/h
```

assigns *speed* to $1 \times m/s$ (which could be written $m/s \times 1$ or $m \times 1/s$ or $1/s \times m$ or just m/s). When the value $5 \times m/s$ is converted to text by *quote*, the result is “5 m/s” without the \times sign and without evaluating the unknown real value *m/s*. And *unquote* “5 m/s” = $5 \times m/s$. Similarly for all units of measurement. One more example: *quote* ($2 \times 3 \times km/h$) = “1.6667 m/s”.

You can define a new unit of measurement, unrelated to the existing units. Measuring unit definition has the form

```
new newname #1
```

The *newname* becomes a unitname. For example,

```
new sheet #1
```

defines a new unit of measurement called the *sheet*. Now you can define the related units

```
new quire:= 25×sheet.
```

```
new ream:= 20×quire
```

You can define a variable using the new units.

```
new order: nat×sheet:= 3×ream
```

This assigns *order* to $1500 \times sheet$. Another example is a monetary unit, such as

```
new dollar #1.
```

```
new cent:= dollar%
```

[Forward Definition](#)

Forward definition has the form

```
new newname
```

The *newname* becomes either a dataname or a programname. For example

```
new mutual
```

is a notice that a definition of *mutual* will follow later in the same scope. In a data definition or program definition, the scope of the name being defined starts with the definition. A forward definition facilitates mutual recursion by starting the scope of a data name or program name even before its definition. For example, leaving gaps for missing parts, in

new $f:=3$. \llbracket **new** f . **new** g (f g \rrbracket . **new** f (f g \rrbracket . \rrbracket
 the inner f and g are each defined in terms of both of them. Without the forward definition of f (following \llbracket), g would be defined in terms of the earlier constant definition **new** $f:=3$.

Name Removal

Names defined with the keyword **new** can become undefined with the keyword **old**. Name removal has the form

old oldname

Ironically, by saying **old** x , the name x becomes available for reuse as a new name. Even though a name becomes undefined, what it named will remain as long as there is an indirect way to refer to it. For example, in predefined dictionary *rand* there are three names *next*, *Int*, and *Real*. They might be defined as:

new *rand* \backslash *var*: 0,..*maxint*:= 123456789. `will be hidden

new *rand* \backslash *next* \llbracket *rand* \backslash *var*:= *mod* (*rand* \backslash *var* \times 5¹³) *maxint* \rrbracket .

new *rand* \backslash *Int* (\langle *from*: *int*. \langle *to*: *int*. *floor* (*from* + (*to*-*from*) \times *rand* \backslash *var*/*maxint*) $\rangle\rangle$ \rrbracket).

new *rand* \backslash *Real* (\langle *from*: *real*. \langle *to*: *real*. *from* + (*to*-*from*) \times *rand* \backslash *var*/*maxint*) $\rangle\rangle$ \rrbracket).

old *rand* \backslash *var*

Variable *rand* \backslash *var* is now hidden; its name is undefined, but *rand* \backslash *next*, *rand* \backslash *Int*, and *rand* \backslash *Real* still use it. And *rand* \backslash *Int* 0 10 has the same value each time it is used until *rand* \backslash *Next* is called. So

rand \backslash *Int* 0 10 + *rand* \backslash *Int* 0 10 = 2 \times *rand* \backslash *Int* 0 10

has value \top . Similarly for *rand* \backslash *Real*.

Synonym Definition

Synonym definition has the form

new newname oldname

The newname becomes a synonym for the oldname. One use is to shorten all names that are deep within several dictionaries. For example, if dictionary *a* contains dictionary *b*, which contains dictionary *c*, which contains dictionary *d*, which contains variable x , then

new x *a* \backslash *b* \backslash *c* \backslash *d* \backslash x

shortens the name *a* \backslash *b* \backslash *c* \backslash *d* \backslash x to just x . The definition

new d *a* \backslash *b* \backslash *c* \backslash d

shortens all names within *a* \backslash *b* \backslash *c* \backslash d , for example, from *a* \backslash *b* \backslash *c* \backslash d x to d x .

Another use is to rename something. To rename *a* to *b*, write

new b *a*. **old** a

The following sequence swaps the names p and q .

new t p . **old** p . **new** p q . **old** q . **new** q t . **old** t

Format

Although not part of the ProTem language, here are some suggested formatting (indentation) rules. The choice of alternative depends on the length of component data and programs.

A. *B*

or *A*.

B

for x : *A* \llbracket *B* \rrbracket

or **for** x : *A*

\llbracket *B* \rrbracket

```

    A || B
or   A
    || B
-----

```

```

    if A [[B]] else [[C]]
or   if A [[B]]
    else [[C]]
or   if A
    [[B]]
    else [[C]]
-----

```

```

or   plan x: A [[B]]
    plan x: A
    [[B]]
-----

```

```

    A + B
or   A
    + B
-----

```

```

    value x: A := B [[C]]
or   value x: A := B
    [[C]]
-----

```

```

    <x: A. <y: B. C>>
or   <x: A. <y: B.
    C>>
-----

```

```

    p [[A]]
or   p
    [[A]]
-----

```

More indentation would show the structure better, but it would crowd programs onto the right side of the page. Consistent indentation improves readability, and is useful redundancy for error checking. (Python uses indentation as part of the syntax; so in Python, indentation is not redundant and cannot be used for error checking.)

Commands

There are 11 commands in ProTem. They are not presented in the grammar, and they cannot be part of a stored program. A command may be given at any time; it does not have to respect the grammatical structure of a program. Each command is the control key combined with a letter. The commands are:

<code>ctl p</code>	pause or resume execution
<code>ctl a</code>	abort execution
<code>ctl v</code>	verify program according to a specification
<code>ctl t</code>	turn type checking on or off
<code>ctl s</code>	stop current session and start new session
<code>ctl u</code>	undo current session
<code>ctl e</code>	enter or exit editor for saved definitions
<code>ctl d</code>	display source or object code for saved definitions
<code>ctl n</code>	display names defined in current scope or persistent scope or in dictionary
<code>ctl m</code>	attach or modify or retrieve memo to defined name
<code>ctl c</code>	generate context comments

Pause

If there is a program being executed, the `ctl p` command pauses its execution. If there is a paused program, `ctl p` resumes its execution.

Abort

Use `ctl a` to abort execution of the currently executing or paused program.

Verify

The command `ctl v` starts a dialogue using *keys* and *screen* to determine the program, bracketed by program brackets `[]` and named by a fancy name, for which verification is wanted. The verification is then attempted. (See [Program Definition](#) and [Named-Program](#).)

Type Checking

Type checking means checking, during execution, that every variable assignment is to a value that is included in the type of the variable. The `ctl t` command asks if you want type checking turned on or off. It then stays that way until the next time you use `ctl t`. Type checking is useful for debugging, but is redundant in a correct program. At the start of a session (see [Session](#)), type checking is on. After verifying (see [Verify](#)), or when we are satisfied that all assignments are always to values included in their types, we can choose not to check types, resulting in faster execution.

Session

Sessions are defined for security and error recovery. When the computer is turned on, a session begins. When some idle time passes (how much time is a parameter of the system and may be set to infinity), a session ends and a new one begins. When the computer is turned off, a session ends. The `ctl s` command causes the current session to end and a new session to begin. If there is a currently executing or paused program, ending the session asks if you want to abort it.

Sessions do not define the lifetime of definitions. A definition in the persistent scope, outside all program bracket pairs `[]`, lasts from the execution of the definition (**new**) to the execution of the corresponding name removal (**old**). This may be less than a session, or more than a session. Turning off the computer should not cut the power instantly, but should first cause the values of any variables in the persistent scope to be saved in nonvolatile memory.

At the start of each session, a programmer must login. This connects the programmer to their persistent scope. Persistent channels (*keys* , *screen* , *microphone* , *speaker* , and *printer*) are initialized and connected to the session. Type checking is turned on. The data definition *session* , which is dependent on channel *keys* , is a text consisting of all keystrokes since the start of the current session. (This is quite practical: an hour of hard work produces only 10 kbytes of keystrokes.) This text can be saved as a record of work done, or for error recovery (see [Undo](#)).

Undo

The command `ctl u` undoes a session (except for inputs and outputs and *session*). Implementing it requires capturing the state at the start of a session. On many computers, returning to the prior state may be cheap; nonvolatile memory (that does not require power) contains the state as it was at the start of the current session, and volatile memory (that requires power) contains the current state. After undo, you can capture the current value of *session* , let us call it *recovery* ,

new recovery: text:= session

then reassign *recovery* , and then execute the result by writing *exec recovery* . This gives us perfectly flexible error recovery for the modest cost of a keystroke file.

[Edit](#)

The edit command `ctl e` is used to modify an existing persistent definition. It invokes a dialogue using *keys* and *screen* to determine which definition. It then invokes an editor. In the editor, `ctl e` exits the editor, and asks if you want to throw away the old definition and save the new definition. If you do want to save it, it is compiled. If you want to create a definition using the editor, first create the definition, for example, `new p [[ok]]`, and then invoke the editor to modify it. If you want to delete a persistent definition, use `old`.

[Display](#)

The command `ctl d` starts a dialogue using *keys* and *screen* to determine the name (simple or compound) of the persistent definition whose source or object code you want to view.

[Names](#)

The command `ctl n` begins a dialogue using *keys* and *screen* to determine whether you want the names defined in the current scope, the persistent scope, or in a (sub)dictionary. In a (sub)dictionary you will see only the first-level names, not the names in its subdictionaries.

[Memo](#)

Each definition can optionally have a memo attached to it. The memo might explain the purpose or use of the definition. It is there to be read by a human, not for execution. A memo is similar to a comment that you would make at the point of definition, but differs in that you can retrieve it anytime. The command `ctl m` starts a dialogue using *keys* and *screen* to determine which name (simple or compound), whether you want to attach a new memo, modify an existing memo, or retrieve an existing memo. For example, you may say that you want to attach the memo

This variable accumulates the sum of the products.

to name *x*. Asking for the memo attached to predefined name *e* prints

e:= 2.718281828459045 (approximately) `constant` The base of the natural logarithms.

[Context](#)

The command `ctl c` starts a dialogue using *keys* and *screen* to determine the program, bracketed by `[[]]`, for which context comments are wanted. The comments are then generated. These comments say which nonlocal names are used, and in what way they are used. Here is the format.

```
`input: on these channels
`output: on these channels
`use: the values of these variables, constants, data names, units, and function names
`assign: these variables
`call: these program names and plan names
`refer: to these dictionaries
```

If there already are comments in this format, they are replaced. For examples of context comments, see [Example Programs](#). Additionally, a programmer may want to include comments like

```
`spec: specification
`pre: precondition
`post: postcondition
`inv: invariant
```

but these are not generated by `ctl c`.

Predefined Names

The predefined names are defined in dictionary *predefined* in the persistent scope (see [Scope](#)). Predefined names can be redefined in a local scope. For example, one of the predefined names is the imaginary number *i* (a square root of -1). You may also want to define a local variable *i*. If you do, you can still refer to the predefined *i* as *predefined*i** (unless you have covered the name *predefined* with a local definition). If predefined name *i* is covered by a definition in a scope outside the local scope where you are working, you can get back the simple name *i* as the imaginary number in these three ways:

```
new i:= predefinedi      `constant definition
new i predefinedi      `synonym definition
new i:= 0&1              `constant definition
```

The command `ctl n` lists the names in the *predefined* dictionary. The command `ctl m` gets a description of a predefined name. The collection of predefined names may change over time.

Each predefined name is one of:

variable	at present, there are no predefined variables
constant	evaluated; not assignable
data	unevaluated; evaluation upon use; not assignable
program	unexecuted; execution upon use
channel	reinitialized at the start of each session
unit	unrelated to other predefined units
dictionary	at present, there is one predefined dictionary <i>rand</i>

Some definitions use § (solutions, those) which is not part of ProTem; it is defined in [a Practical Theory of Programming](#). Here are the predefined names.

abs: com→*real* data Absolute value. $abs\ x = \sqrt{re\ x^2 + im\ x^2}$.

*all = com, char, bin, /all, *all, [all]* data All ProTem data values.

arc: com → § $\langle r: real. 0 \leq r < 2 \times \pi \rangle$ data The angle or arc of a complex number.

arccos: § $\langle r: real. -1 \leq r \leq +1 \rangle$ → § $\langle r: real. 0 \leq r \leq \pi/2 \rangle$ data A trigonometric function.

arcsin: § $\langle r: real. -1 \leq r \leq +1 \rangle$ → § $\langle r: real. 0 \leq r \leq \pi/2 \rangle$ data A trigonometric function.

arctan: real → § $\langle r: real. 0 < r < \pi/2 \rangle$ data A trigonometric function.

await program A plan with one constant parameter of type *real**xs*. If the argument represents the present or a future time, its execution does nothing but takes time until the instant given by the argument. If the argument represents the present or a past time, its execution does nothing and takes no time. See *time* and *wait* and *s*.

bin:= T, ⊥ constant The binary values.

bold: text→*text* data Same text but in bold font.

capital: char constant The English capital letters. See *small* and *letter*.

ceil: real→*int* data $r \leq \text{ceil } r < r+1$

char data The characters.

charnat: char→*nat* data A one-to-one function with inverse *natchar*. The encoding might be extended ASCII or unicode. Character combinations, for example shift-option-a, also have numeric encodings.

click: char constant The click character.

com = real&*real* data The complex numbers.

cos: real → § $\langle r: real. -1 \leq r \leq +1 \rangle$ data The trigonometric cosine function.

cosh: com→*com* data The hyperbolic cosine function.

cursor: *nat*; *nat* *data* A data name whose value is the current cursor position.

definition: *text*→*text* *data* If the argument is the name of a persistent definition, then the result is the textual definition. For example, if *combo* is defined as **new** *combo*:= “10-20-30” , then *definition* “*combo*” = “*combo*:= “10-20-30”” . Otherwise the result is the text “undefined” .

delete: *char* *constant* The delete or backspace character.

digit:= “0”, “1”, “2”, “3”, “4”, “5”, “6”, “7”, “8”, “9” *constant* The decimal digits.

div: *real* → §{*r*: *real*. *r*>0} → *int* *data* *div a d* is the integer quotient when *a* is divided by *d* .
 $(0 \leq \text{mod } a \ d < d) \wedge (a = \text{div } a \ d \times d + \text{mod } a \ d)$

doubleclick: *char* *constant* The doubleclick character.

drain *program* A plan with one input channel parameter. Reads and discards any unread input.

e:= 2.718281828459045 (approximately) *constant* The base of the natural logarithms.

end: *char* *constant* The end character. It has zero width. It is greater than all other characters.

even: *int*→*bin* *data* A function that says whether its argument is even.

exec *program* A plan with one text parameter. If the argument represents a ProTem program, the execution is that of the represented program. It “unquotes” its argument. If applied to “*x*:= *x*+1” , the “*x*” refers to whatever *x* refers to at the location where *exec* “*x*:= *x*+1” occurs. If the argument does not represent a ProTem program, execution displays an error message on *msg* . To evaluate data represented as text, for example “2+2” , use *unquote* .

exp: *com*→*com* *data* The exponential function. *exp x* = e^x .

false:= ⊥ *constant* A binary value.

find: *all*→*all*→*nat* *data* If *i* is an item in string *S* , then *find i S* is the index of its first occurrence; if not, then *find i S* = $\leftrightarrow S$.

fit: *int*→*text*→*text* *data* If $i \geq 0$ then *fit i t* is a text of length *i* obtained from *t* either by chopping off excess characters from the right end or by extending *t* with spaces on the right end. If $i \leq 0$ then *fit i t* is a text of length $-i$ obtained from *t* either by chopping off excess characters from the left end or by extending *t* with spaces on the left end.

floor: *real*→*int* *data* *floor r* $r \leq r < 1 + \text{floor } r$

g *unit* A mass of one gram.

i:= 0&1 *constant* An imaginary number: a square root of -1 .

im: *com*→*real* *data* The imaginary part of a complex number.

infinity:= ∞ *constant* An infinite number, greater than all other numbers.

int = *nat*, $-nat$ *data* The integers.

italic: *text*→*text* *data* Same text but in italic font.

keys? *char!* “” *channel* To the program that monitors key presses, it is an output channel; to all other programs, it is an input channel.

lb: §{*r*: *real*. *r*>0} → *real* *data* The binary (base 2) logarithm.

letter:= *small*, *capital* *constant* The English small and capital letters.

ln: §{*r*: *real*. *r*>0} → *real* *data* The natural or Napierian (base *e*) logarithm.

log: §{*r*: *real*. *r*>0} → *real* *data* The common (base 10) logarithm.

m *unit* A distance of one meter.

mail *channel* A text input and output channel for email.

match: *all*→*all*→*nat* *data* If *pattern* occurs within *subject* , then *match pattern subject* is the index of its first occurrence. If not, then *match pattern subject* = $\leftrightarrow \text{subject}$.

maxint: *int* *constant* The maximum representable integer (machine dependent).

maxreal: *real* *constant* The maximum representable real number (machine dependent).

microphone? **sound!* *silence* *channel* To the microphone, it is an output channel; to all other programs, it is an input channel.

minint: *int* *constant* The minimum representable integer (machine dependent).

mod: *real* → §{*r*: *real*. *r*>0} → *real* *data* *mod a d* is the remainder when *a* is divided by *d* .
 $(0 \leq \text{mod } a \ d < d) \wedge (a = \text{div } a \ d \times d + \text{mod } a \ d)$

movie = **picture* data All strings of pictures.

msg? text! “” channel A channel for displaying messages from the ProTem implementation.

nat = 0,..∞ data The natural numbers.

natchar: nat→*char* data A one-to-one function with inverse *charnat* . The encoding might be extended ASCII or unicode. Character combinations, for example shift-option-a, also have numeric encodings.

nil constant The empty string.

nl: char constant The new line or next line or return or enter character.

null constant The empty bunch.

numpat: text constant A text pattern for reading a number from a text channel.

odd: int→*bin* data A function that says whether its argument is odd.

ok program A program whose execution does nothing and takes no time.

ord = *real*, *char*, *bin*, *fall*, **ord*, [*ord*] data The ordered type, for which \wedge \vee $<$ $>$ \leq \geq are defined.

pi:= 3.141592653589793 (approximately) constant The ratio of a circle's circumference to its diameter.

picture = [*x**[*y**(0,..*z*)]] data where *x* is the number of pixels in the horizontal dimension, *y* is the number in the vertical dimension, and *z* is the number of pixel values.

plain: text→*text* data Same text but not italic, not bold, and not underlined.

point data A function that applies to a list and gives its deep domain (a bunch of strings of indexes). It is a signal to the implementation that the strings in it will be used only as indexes to the list. It can therefore be implemented as a memory address (pointer).

pre: char→*char* constant The character predecessor function. *pre* “b” = “a” ; *pre* “ ” = “ ”

printer? text! “” channel To the printer, it is an input channel; to all other programs, it is an output channel.

ProTem: text constant This document.

quote: all→*text* data produces a text representation of its argument. The argument is evaluated before quoting. See *unquote* and *realtxt* .

rand \ dictionary containing three definitions.

next program Assigns a hidden variable to the next value in a random sequence.

Int: int→*int*→*int* data A function that is dependent on a hidden variable, and is reasonably uniform over the interval from (including) the first argument to (excluding) the second argument.

Real: real→*real*→*real* data A function that is dependent on a hidden variable, and is reasonably uniform over the interval between the arguments.

rat = *int*/(*nat*+1) data The rational numbers.

re: com→*real* data The real part of a complex number.

real data The real numbers including ∞ and $-\infty$.

realtxt: nat→*nat*→*nat*→*real*→*text* data Format a real number. *realtxt d e w r* is a text representing real *r* with the final digit rounded. *d* is the number of digits after the decimal point; if *d*=0 the point is omitted. *e* is the number of digits in the exponent; if *e*>0 the decimal point will be placed after the first significant digit; if *e*=0 the ^^ is omitted and the decimal point will be placed as necessary. *w* is the total width; if *w* is greater than necessary, leading blanks are added; if *w* is less than sufficient, the text contains blobs.

realtxt 4 1 10 *pi* = “ 3.1416^^0” *realtxt* 2 0 6 (-*pi*) = “ -3.14”

realtxt 0 0 3 5 = “ 5” *realtxt* 0 0 3 (-5) = “ -5” *realtxt* 0 0 2 123 = “••”

See *quote* .

round: real→*int* data $r-0.5 \leq \text{round } r < r+0.5$

s unit A time of one second.

screen? text! “” channel To the screen, it is an input channel; to all other programs, it is an output channel.

session: *text* data The join of all texts from channel *keys* since the start of a session.

sign: *real* → (-1, 0, 1) data The sign of a real number.

silence: *sound* data The silent sound.

sin: *real* → §{*r*: *real*. -1 ≤ *r* ≤ +1} data The trigonometric sine function.

sinh: *com*→*com* data The hyperbolic sine function.

small: *char* constant The English small letters. See *capital* and *letter* .

sort program A plan with one variable parameter of type **ord* . Sorts in nondecreasing order.

sound data The sounds.

speaker? **sound!* *silence* channel To the speaker, it is an input channel; to all other programs, it is an output channel.

sqrt: *com*→*com* data The principal square root. $4^{(1/2)} = 2, -2$ but *sqrt* 4 = 2 .

stop program Its execution does nothing and takes forever so that no computation can follow.

sub: *all*→*nat*→*nat*→*all*→*all* data *sub s n m t* is a string formed from string *s* by replacing the substring from index *n* to index *m* with string *t* . The substring being replaced *s_(n;..m)* does not have to be the same length as the string *t* replacing it. If *n=m* this is insertion. If *t=nil* this is deletion. *sub s n m t* = *s_(0;..n)*; *t*; *s_(m;..↔s)*

subst: *all*→*all*→*all*→*all* data *subst s x y* is a string formed from string *s* by replacing all occurrences of item *x* with item *y* .

suc: *char*→*char* constant The character successor function. *suc* “a” = “b” ; *suc end* = *end*

tab: *char* constant The tab character.

tan: *real*→*real* data The trigonometric tangent function.

tanh: *com*→*com* data The hyperbolic tangent function.

text = **char* data The character strings.

texttime: *text*→(*realxs*) data If the argument represents a time, possibly preceded by space, tab, and new line characters, possibly followed by space, tab, and new line characters, the result is the represented time in seconds since 2000 January 1 at 0:00 UTC (the midnight that begins 2000 January 1 at longitude 0). Times before then are negative. For example, *texttime* “1947 September 16 at 14:24:32.5 UTC-5” = -68675727.5*xs* . Otherwise the result is -∞*xs* . See *timetext* .

time: *realxs* data A data name whose value depends on a clock. It gives the current time in seconds since 2000 January 1 at 0:00 UTC (the midnight that begins 2000 January 1 at longitude 0). Times before then are negative.

timetext: (*realxs*)→(-12,..12)→*text* data Given the time in seconds since 2000 January 1 at 0:00 UTC (the midnight that begins 2000 January 1 at longitude 0), and a time zone, the result is a readable text. Times before then are negative. See *texttime* . For example, *timetext* (-68675727.5*xs*) (-5) = “1947 September 16 at 14:24:32.5 UTC-5”

trim: *text*→*text* data A text formed from the argument by removing all leading and trailing space, tab, and new line characters.

true:= T constant A binary value.

underline: *text*→*text* data Same text but underlined. Any part of the text that is already underlined is underlined again.

unquote: *text*→*all* data If the argument represents a ProTem data expression, the result is the value of the represented data. The argument is evaluated after unquoting. If the argument does not represent a ProTem data expression, the function is not evaluated. See *quote* .

wait program A plan with one constant parameter of type *realxs* . If the argument is nonnegative, its execution does nothing and takes the time given by the argument. If the argument is nonpositive, its execution does nothing and takes no time. See *await* and *time* and *s* .

Miscellaneous

The ProTem equivalent of enumerated type is shown here.

```
new brush: “red”, “green”, “blue” := “red”
or new color: := “red”, “green”, “blue”.
new brush: color := “red”
```

The ProTem equivalent of the record type (structure type) is as follows.

```
new p: “name” → text | “age” → nat := “name” → “Josh” | “age” → 16
or new person: := “name” → text | “age” → nat.
new p: person := “name” → “Josh” | “age” → 16
```

The fields of p can be selected by data that evaluates to text, for example

```
p “name”
is the text “Josh”. The value of p can be changed using a function arrow and selective union.
p := “name” → “Amanda” | “age” → 2.
p := “age” → 3 | p
```

We can even have a whole file (string) of records and join new records onto its end.

```
new file: *person := nil.
file := file; p
```

When the predefined function *point* is applied to a list argument, it yields the deep domain of the list. For example,

```
point [10; [11; 12]; 13] = 0, 1;(0, 1), 2 = 0, 1;0, 1;1, 2
```

When used as a type or as part of a type, *point* is unusual; it does not quite obey the language rules. For example, we can define a linked list G as follows.

```
new G: [* (“name” → text | “next” → point  $G$ )] := [“name” → “Alice” | “next” → 0].
new current: point  $G$  := 0
```

Contrary to the rule, the type mentions G , the variable being defined. The occurrences of *point* G are the deep indexes of G at all times, not just at the time the definition is executed. The use of *point* is a signal to the implementation that its strings of natural numbers will be used only as indexes into G (and the implementation will check that this is so). Therefore they can be implemented as memory addresses, giving us the efficiency of pointers. The initial value of G is a list of length 1, so initially the only possible value of G 0 “next” is 0, and the only possible value of *current* is 0. Now suppose G is reassigned as follows:

```
G := [“name” → “Bob” | “next” → 1];;
```

Then G has length 2, so G 0 “next” can have value 0 or 1. Similarly *current* can have value 0 or 1, so the assignment

```
current := current + 1
```

can now be made, but the assignment *current* := *current* + 2 cannot be made at this time. It is possible that an assignment to G may make the values of G 0 “next” and *current* illegal, which is a flaw in this use of *point* known as a “dangling pointer” or “dangling reference”. This use of *point* is unsafe, and that is the price for pointer efficiency.

The previous example, with linked list G , does not show the full generality of *point*. Here is a tree-structured example.

```
new t: tree ([nil], [tree; nat; tree]) := [nil].
new p: point  $t$  := nil
```

If tree t gains some branches, for example,

```
 $t$  := [[[nil]; 2; [[nil]; 5; [nil]]]; 3; [[nil]; 7; [nil]]]
```

we can move p down to the left in the tree with the assignment

$p := p; 0$
and move it down to the right with the assignment

$p := p; 2$

Thus p is a string of indexes indicating a subtree $t@p$ of t . We can replace this subtree with tree s using the assignment

$t := p \rightarrow s \mid t$

We can express the information at the node indicated by p as

$t@p \ 1$ or $t@(p; 1)$

and we can replace the information at this node with the integer 6 using the assignment

$t := (p; 1) \rightarrow 6 \mid t$

To move up in the tree, we just remove the final item of p

$p := p_{-}(0; .. \leftrightarrow p-1)$

The “procedure”, “method”, or “function” of some other programming languages is a combination of naming, scope, and parameter(s). For example,

new transform \llbracket **plan** *magnification: real* \llbracket **plan** *translation: real*
 $\llbracket x := magnification \times x + translation \rrbracket \rrbracket$

Here is a definition of a plan with one parameter

new translate $\llbracket transform \ 1 \rrbracket$

formed by providing one argument to a two-parameter plan. To provide an argument for just the second parameter is a little more awkward, but not too bad.

new magnify \llbracket **plan** *magnification: real* $\llbracket transform \ magnification \ 0 \rrbracket \rrbracket$

We can now obtain a three-times magnification of x in either of these ways:

magnify 3 0 or *transform 3 0*

In some other programming languages, the “function” is a combination of naming, scope, parameter(s), and **value**-data. For example,

new factorial $\langle \langle n: nat. \ value \ f: nat := 1 \llbracket for \ i: 1; ..n+1 \llbracket f := f \times i \rrbracket \rrbracket \rangle \rangle$

Exception handling is provided by \mid or **if** or $\models \Rightarrow$. For example,

new divide $\langle \langle dividend: com. \ \langle divisor: com. \ divisor = 0 \models \text{“zero divide”} \Rightarrow dividend / divisor \rangle \rangle \rangle$

Then

divide: com $\rightarrow com \rightarrow (com, \text{“zero divide”})$

The selective union operator applies its left side to an argument if that argument is in the stated domain of its left side; otherwise it applies its right side. Let us define

new weekday $:= \langle d: 0; ..7. \ 1 \leq d \leq 5 \rangle$

Then if i fails to be an integer in the range $0; ..7$ in the expression

$(weekday \mid all \rightarrow \text{“domain error”}) \ i$

the left side of \mid “catches” the exception and “throws” it to the right side, where it is “handled”.

Input choice, as in CSP, can be programmed round-robin as follows.

inputchoice \llbracket **if** $c?? \llbracket c? \ nil \ (\text{numpat}) \ end. \ P \rrbracket$
else \llbracket **if** $d?? \llbracket d? \ nil \ (\text{numpat}) \ end. \ Q \rrbracket$
else $\llbracket inputchoice \rrbracket \rrbracket$

In the persistent scope, ProTem functions as an operating system, where programs are executed as soon as they are entered. Unix directories are dictionaries. Unix files are variables. Unix cp is an assignment. Unix rm is ProTem's **old**. Unix mv is a synonym definition followed by **old**. Unix ls and man commands are $\llbracket \text{ctl } n \rrbracket$ and $\llbracket \text{ctl } m \rrbracket$. The effect of Unix pipes is obtained by channel parameters. For example, suppose *trimmer* is a plan to trim off leading and following blanks and tabs from lines

of text, and *enumerate* is a plan to number lines.

```
new trimmer [[plan in? text [[plan out! text
                [[loop [[if in?? [[in? nil (text) nl. out! trim (in?); nl. loop]]]]]]]].
new enumerate [[plan in? text [[plan out! text
                [[new n: nat:= 0.
                  loop [[if in?? [[in? nil (text) nl. out! n; “ ”; in?; nl. n:= n+1. loop]]]]]]]]]
```

We can feed the output from *trimmer* to the input of *enumerate* by defining a channel for the purpose. If the original input comes from *keys*, and the final output goes to *screen*, then

```
new pipe? text! “”. trimmer keys pipe. enumerate pipe screen. old pipe
```

Even better:

```
new pipe? text! “”. trimmer keys pipe || enumerate pipe screen. old pipe
```

If *enumerate* wants input before it is available from *trimmer*, *enumerate* waits. If *trimmer* sends output before *enumerate* wants it, it is buffered.

Unix mail is ProTem's *mail* channel. If you are the creator of the definition of *something* in the persistent scope, and you want to send it to *someone* for them to make changes, then

```
mail! “To: someone@address.domain”; nl; definition “something”
```

(see predefined *definition*). When *someone* sends back the changed definition, receive it, delete your old definition, and then redefine it (see predefined *exec*) by

```
mail? “” (text) end. old something. exec (mail?)
```

An implementation may provide plans for a variety of languages. For example, it may provide a plan named *Python*, with one text parameter, whose execution is that of the Python fragment represented by the argument. It may provide *asm*, whose execution is that of the assembly-language program represented by the argument.

ProTem considers object orientation to be a programming style, rather than a programming-language style, or collection of language features. Object-oriented programming (as a style of programming) can be done in ProTem. Data structures, and the functions and procedures that access and update them, can be defined together in one dictionary. If many “objects” of the same type are wanted, new dictionaries just like old ones are easily defined using `\` (see *parseStack* in [Dictionary Definition](#)).

To execute a program stored on someone else's computer, just invoke that remote program using its full address (computername and programname). For efficiency, it might be best to compile that remote program for your own computer and run it locally. Any nonlocal names (variables, channels, and so on) refer to entities on the computer where the program is compiled.

I would prefer to use the symbols \uparrow and \downarrow for maximum and minimum as in [unified algebra](#), rather than \vee and \wedge . But \uparrow and \downarrow are not the tradition for binary disjunction and conjunction, nor for set union and intersection, and there are no good keyboard substitutes for them. As it is, the ordering symbols $\vee \wedge < > \leq \geq$ are nicely related.

In ProTem, a text is written with quotation marks. An attractive alternative to quotation marks is underlining. For example, the text “abc” would instead be written abc. It is a good-looking syntax, but awkward to key in, so quotation marks could be the keyboard substitute. (The empty text would be *nil*, string indexing would not be underscore, and *quote* and *unquote* would be renamed *textify* and *eval*.) I have used underlining in ProTem just for quotation marks within a text; for example, “Just say “no”.”. This would become Just say “no”., with no exception needed for quotation marks within a text. In ProTem, if you want an underlined quotation mark within a text, you have to underline it again. And so on. Thus every character can occur within a text. In the

alternative notation, no special rule is needed. Underlining presents a theoretical (but not practical) limitation: we cannot write a self-reproducing text (try). We can write a self-reproducing text with quotation marks, repeating the left and right quotation marks as characters within a text. For example, “Just say ““no”””. Using this convention, here is a self-reproducing text:

““““(0;0;(0;..32);31;31;(1;..31))”””“(0;0;(0;..32);31;31;(1;..31))””””

Perform the indexing to see what you get.

In the [Presentation Grammar](#), the **case**-program has the form

case data \llbracket program \rrbracket

with the program form

program \llbracket program

to make more cases. Although it is correct, it is not a clear presentation of

case data \llbracket program \rrbracket \llbracket program \rrbracket \llbracket program \rrbracket

and so on. A clear presentation requires another nonterminal, as in the [LL\(1\) Grammar](#) and [LR\(0\) Grammar](#). My principle is that the [Presentation Grammar](#) has only two nonterminals (program and data), but the **case**-program stretches that principle rather thin. Furthermore, the program form

program \llbracket program

is the only program or data form whose use is confined to a single context, and I am unhappy to have any program or data form that cannot be used more generally. If I were to allow more nonterminals, the cases could be labeled, so we needn't count them.

There is both a one-tailed **if** and a two-tailed **if** in ProTem, but there is no dangling-**else** problem.

Sounds and pictures are data structures. This part of ProTem is not yet designed. Perhaps a picture is an element of $[x*[y*(0,..z)]]$ where x is the number of pixels in the horizontal direction, y is the number of pixels in the vertical direction, and z is the number of pixel values. A picture could therefore be expressed in the same way as any other two-dimensional array, and one could refer to the pixel in column 3 and row 4 of picture p as $p\ 3\ 4$. Perhaps a movie is a string of pictures. The operations on movies would be those of strings, such as substring and join. To help in the creation of movies, one of the pixel values should be transparent, and one of the operations on pictures should be overlaying. Predefined *silence* is a sound, and predefined *sound* is all sounds. Sounds are input on channel *microphone*; pictures are input on channel *camera*. A constant can be defined as a sound or picture. A variable can be assigned to a sound or picture. Sounds and pictures can be included in a data structure, and manipulated using the operators on that data structure. Sounds can be output on channel *speaker*. Channel *screen* must be modified so pictures can be output on channel *screen*.

Intentionally Omitted Features

Each of the following omitted features would be a small syntactic convenience, and would be easy to add to the language. But they would make the language larger, and that would be a cost. And they would move away from the form needed for verification. So they are not included in ProTem.

assertion

assert $x \leq y$ means **if** $-(x \leq y)$ \llbracket ! “assert failure”. *stop* \rrbracket

name grouping

new $x, y: int := 0$ means **new** $x: int := 0$ **||** **new** $y: int := 0$

old x, y means **old** x **||** **old** y

$x, y := 0$ means $x := 0$ **||** $y := 0$

$\langle a, b: nat. a+b \rangle$ means $\langle a: nat. \langle b: nat. a+b \rangle \rangle$

plan $a, b: nat \llbracket x := a+b \rrbracket$ means **plan** $a: nat \llbracket$ **plan** $b: nat \llbracket x := a+b \rrbracket \rrbracket$

item assignment

$S_3 := 5$	means	$S := S \leftarrow 3 \triangleright 5$
$L_3 := 5$	means	$L := 3 \rightarrow 5 \mid L$
$A_3_4 := 5$	means	$A := (3;4) \rightarrow 5 \mid A$

loop constructs

while $n > 0$ $[[n := n - 1]]$	means	<i>while</i> $[[\text{if } n > 0 \text{ } [[n := n - 1. \text{ while}]]]]$
repeat $[[n := n - 1]]$ $n = 0$	means	<i>repeat</i> $[[n := n - 1. \text{ if } \neg(n = 0) \text{ } [[repeat]]]]$
loop $[[n := n - 1. \text{ if } n = 0 \text{ } [[\text{exit } 1]]]. \text{ loop } [[m := m - 1. \text{ if } m = 0 \text{ } [[\text{exit } 2]] \text{ else } [[\text{exit } 1]]]]]$	means	<i>loop</i> $[[n := n - 1. \text{ if } \neg(n = 0) \text{ } [[m := m - 1. \text{ if } \neg(m = 0) \text{ } [[loop]]]]]]$

return from named-program

<i>name</i> $[[\text{some. if } n = 0 \text{ } [[\text{return } name]]. \text{ more}]]$	means	<i>name</i> $[[\text{some. if } \neg(n = 0) \text{ } [[more]]]]$
---	-------	--

The assignment $L := 3 \rightarrow 5 \mid L$ should be compiled the same as $L_3 := 5$ would be if it were included in ProTem; the list L should not be copied. The same for string item assignment. In the loop *while* $[[\text{if } n > 0 \text{ } [[n := n - 1. \text{ while}]]]]$

the last-action (tail recursive) call should be compiled as a branch instruction, with no stack activity, the same as a **while**-loop would be if it were included. The same for other loop constructs. Omitting item assignment and looping constructs should not cost execution time or space.

Plans with parameters and arguments of type program

plan <i>simpname</i> $[[\text{program}]]$	plan, parameter is program
<i>program</i> <i>program</i>	plan, program argument

were considered and rejected due to syntactic and semantic ambiguities.

As a counterpart to the Unix `cd` command, I considered

open *dictionaryname*
close *dictionaryname*

to allow names in that dictionary to be referred to without stating the dictionary. For example, if we have dictionary *abc*, and within it names *x* and *y*, we refer to these names as *abc**x* and *abc**y*. By saying **open** *abc* we can then refer to them as just *x* and *y*. But the interaction between **open** and scope is complex, we can already refer to names within *predefined*, and we can shorten names by synonym definition, so I left out **open** and **close**.

The syntax

new *abc**def*

defines a new dictionary *abc* populated with the definitions from existing dictionary *def*. Perhaps it should instead add the definitions of existing dictionary *def* to existing dictionary *abc*. At present, there is no way to do that. If I were to make this change, there would still be a way to define a new dictionary populated with definitions from an existing dictionary:

new *abc*\. **new** *abc**def*

But I would have to decide how to resolve duplicate definitions. I am guessing that we always want to populate a new dictionary, not further populate an existing one.

There is no frame construct in ProTem, but `[ctl c]` serves a similar purpose. In some languages there is a module or object construct for the purpose of grouping together related definitions. In ProTem, dictionaries serve that purpose.

In order to allow sharing of variables in the persistent scope, I created a personal identity type and a scheme of permits to say who can use and change what. The scheme was more flexible than Unix `chmod`, and I was quite pleased with it. But with sharing, all mathematical reasoning is invalid; $x = x$

might evaluate to \perp . And with sharing, locks are required so that assignments do not interfere with uses, and assignments do not interfere with other assignments. Sharing problems are solved by the use of channels: anyone can email anything to anyone else; any program can send anything to any other program on a channel. So, even though it hurts to leave out something I worked hard on, identities and shared variables are intentionally omitted.

Two binary operators Δ (nand) and ∇ (nor) are missing. They are not wanted very often, there are no good keyboard substitutes for them, and they are easily synthesized:

$$x\Delta y = \neg(x\wedge y) \quad x\nabla y = \neg(x\vee y)$$

The empty program, denoted by nothing, whose execution does nothing and takes no time, is easy to add to ProTem. With it, we can almost always consider the period (.) to be a program terminator, rather than an infix connective. But there is one context where it causes confusion.

$A. \parallel B.$

looks like program A (terminated with a period) is in parallel with program B (terminated with a period). But, according to the order of execution, the empty program is in parallel with B , as in

$A. [\parallel B]$.

So I have not included the syntactically empty program in ProTem, providing instead the predefined program *ok* whose execution does nothing and takes no time (as in [aPTOP](#)).

I am tempted to omit both

if data $[\text{program}]$

case data $[\text{program}]$

and insist that both **if** and **case** always have an **else**, but so far they remain in ProTem.

Implementation Philosophy

Some expressions do not need to be evaluated, and some cannot be evaluated. For examples,

! -3	\prints -3
! [0; 1] 2	\should print [0; 1] 2
! [0; 1] 2 = [0; 1] 2	\should print \top
! 4^(1/2)	\should print 2, -2
! 1/0	\should print 1/0
! 0/0	\should print 0/0
! 1/0 = 1/0	\should print \top
! <r: rat. 5> (1/0)	\should print 5

ProTem does not evaluate the application of the negation operator \neg to the operand 3 (see [Number Representation](#)); it just prints the operator and operand. Similarly other expressions that cannot be evaluated should just print the expression. Due to the difficulty of implementation, it is permissible for an implementation to behave differently.

No programming language has ever been, nor will ever be, implemented entirely. Every programming language is infinite; every implementation is finite. There is always a program too big for the implementation. There is a multitude of size limitations: the parse stack might overflow, the dictionary (symbol table) might be too small, the forward branch fixup list might be exceeded, and so on. It would be ugly to define a programming language by listing all the size limitations of programs. And it would be counter-productive because it would exclude implementations that can accommodate larger programs.

Whenever a program exceeds a size limitation, the implementation should not say “Error: limitation exceeded.”, because the program is not in error. The implementation should say “Apology: this implementation is too limited to accommodate your program.”. An “error” message tells a programmer to correct the error; there is no other option. An “apology” message gives the programmer 3 options: change the program to live within the limitation; change the implementation options to increase the limit that was exceeded; take the program to a different implementation.

Natural numbers and integers are usually limited to those that are representable in a specific number of bits, for example, 32 bits. This is a size limitation, just the same as other size limitations. It is more complicated and uglier to define arithmetic within finite limitations than to define the naturals and the integers. And it is counter-productive to do so, because it excludes an implementation with 64-bit arithmetic. As with other implementation limitations, numeric overflow should not get an “error” message; it should get an “apology” message.

Floating-point numbers and arithmetic should never be offered as a language feature. The programmer wants rational or real numbers and arithmetic, but may be willing to accept the floating-point approximation for the sake of efficiency. Floating-point, with a specific number of bits, is an implementation limitation. Any [alternative](#) to floating-point that increases the accuracy without taking too much time or space should be welcome.

ProTem is a rich programming system, offering many kinds of data and operators on data, and many ways to structure a computation. Some features may be difficult to implement. And some features may be of little use to most programmers. It may be a wise decision not to implement some features. For example, an implementer might decide that in a variable definition, the type must be one of

*int rat bin text [n*type]*

where *n* is a natural number and *type* is any of these types just listed. An implementer may decide not to implement concurrent execution. No-one can complain that the complete language is not implemented, since it is impossible to completely implement any language. But ProTem is defined to allow all type expressions that make sense, and to allow concurrency, so the next implementation can accommodate programs that previous implementations could not accommodate.

Example Programs

Portation Simulation

new *simport* ` a program to simulate [portation](#)

```
[[ `input: keys
  `output: screen
  `use: ceil m nat nil nl point real s sqrt time
  `call: await stop
  `refer: rand
```

- ` Distance between control boxes is always 1 m.
- ` Merges do not overlap, so there is at most 1 corresponding box on the merging portway.
- ` Each divergence has a left branch and a right branch; there is no straight.
- ` Leading to a divergence, boxes record only one square speed.

` start of definitions

new *km*:= 1000×*m*. **new** *h*:= 60×60×*s*. ` kilometer and hour

new *maxaccel*:= 1.5×m/s/s. ` maximum deceleration = −*maxaccel*
new *speedlimit*:= 60×km/h. ` speed limit is 60 km/h everywhere
new *cushion*:= 1×s. ` reaction time for all porters
new *impatience*:= 10/s. ` acceleration factor
new *maxdistance*:= *ceil* (*speedlimit*² / (2×*maxaccel*)). ` max search distance ahead
new *numporters*:= 120. **new** *numboxes*:= 7480.
new *visualDelayTime*:= 0.5×s. ` for human viewing

new *porter*. ` so *porter* can be referenced before it is defined

new *box*: [*numboxes* * ((“ahead left”, “ahead right”, “behind left”, “behind right”) → *point box*
 | “beside” → *point box*
 | “above” → (*point porter*, *numporters*)
 | (“horizontal”, “vertical”) → *nat*)] ` box position on screen
 := [*numboxes* * ((“ahead left”, “ahead right”, “behind left”, “behind right”) → 0
 | “beside” → 0
 | “above” → *numporters* ` indicates no porter above
 | (“horizontal”, “vertical”) → 0)].

new *porter*: [*numporters* * (“below” → *point box* ` what is beneath
 | “arrival time” → (*real*×s) ` arrival time at this box
 | “speed” → (*real*×m/s))] ` current speed
 := [*numporters* * (“below” → 0
 | “arrival time” → (0×s)
 | “speed” → (0×m/s))].

new *draw* [[**plan** *b*: *nat* [[**plan** *c*: “grey”, “blue”, “red” [[UNFINISHED]]]]. ` end of *draw*
 ` draws a box at screen position (*box b* “horizontal”) (*box b* “vertical”) of color *c*.
 ` “grey” means no porter present, “blue” means porter present, “red” means crash
 ` UNFINISHED because graphical output has not yet been designed

` end of definitions, start of initialization

for *b*: 0;..*numboxes*

[[! “What box is ahead-left of box ”; *b*; “?” . ?!.
box:= (*b*; “ahead left”) → ? | (?; “behind left”) → *b* | *box*.
 ! “What box is ahead-right of box ”; *b*; “?” . ?!.
box:= (*b*; “ahead right”) → ? | (?; “behind right”) → *b* | *box*.
 ! “What box is beside box ”; *b*; “?” . ?!.
box:= (*b*; “beside”) → ? | *box*.
 ! “What are the horizontal and vertical coordinates of box ”; *b*; “?” .
 ?!. *box*:= (*b*; “horizontal”) → ? | *box*.
 ?!. *box*:= (*b*; “vertical”) → ? | *box*.
draw b “grey”]]. ` default color; may be changed below

for *p*: 0;..*numporters*

[[! “Porter ”; *p*; “ is over what box? ”. ?!.
porter:= (*p*; “below”) → ? | *porter*. *box*:= (?; “above”) → *p* | *box*.
draw (?) “blue”]].

⋄ end of initialization, start of simulation

infiniteLoop

[[**new** *iterationStartTime* := *time*. ⋄ the time of the start of each iteration of the *infiniteLoop*
new *p*: *point porter* := 0. ⋄ *p* := the porter that arrived at its current position first
new *t*: *real* × *s* := ∞ × *s*. ⋄ *t* is a time, initially an infinite time
for *q*: 0; .. *numporters* [[**if** *porter q* “arrival time” < *t* [[*t* := *porter q* “arrival time”. *p* := *q*]].
old *t*.

new *b* := *porter p* “below”. ⋄ the box below porter *p*
new *bb* := *box b* “beside”. ⋄ the box beside *b*; if none then *bb* = *b*
new *boxesToDo*: *[*point box*; *nat* × *m*] := *nil*.
⋄ queue of boxes to be explored; their distances ahead of porter *p*
⋄ queue is sorted by increasing distance ahead
⋄ difference between any two distances in the queue is at most 1

⋄ initialize *boxesToDo*

if *bb* = *b* [[*boxesToDo* := *nil*]]
else [[**if** *box bb* “above” = *numporters* [[*boxesToDo* := *nil*]]
else [[**if** *porter (box bb* “above”) “speed” < *porter p* “speed” [[*boxesToDo* := *nil*]]
else [[*boxesToDo* := [*bb*; 0 × *m*]]]].
boxesToDo := *boxesToDo*; [*box b* “ahead left”; 1 × *m*].
if *box b* “ahead left” ≠ *box b* “ahead right” [[*boxesToDo* := *boxesToDo*; [*box b* “ahead right”; 1 × *m*]]].
old *b*. **old** *bb*.

new *accel*: *real* × *m* / *s* / *s* := *maxaccel*. ⋄ acceleration for porter *p*

⋄ using *boxesToDo* calculate *accel* for porter *p*

nextBox [[**new** *b* := (*boxesToDo*₀) 0. ⋄ the box we are looking at
new *d* := (*boxesToDo*₀) 1. ⋄ its distance ahead of porter *p*
boxesToDo := *boxesToDo* (1; .. ↔ *boxesToDo*).
if *d* ≤ *maxdistance*
[[**new** *desiredspeed* ⋄ according to porter *pa*
((*pa*: *point porter*, *numporters*.
pa = *numporters* ≡ *speedlimit*
⇒ (*sqrt* (*porter pa* “speed”² + 2 × *maxaccel* × *d*
+ (*maxaccel* × *cushion*)²)
− *maxaccel* × *cushion*) ∧ *speedlimit*)].
accel := ((((*desiredspeed* (*box b* “above”)
∧ *desiredspeed* (*porter (box b* “beside”) “above”)
− *porter p* “speed”)
× *impatience*)
∨ −*maxaccel*) ∧ *maxaccel*.
if *box b* “above” = *numporters* = *porter (box b* “beside”) “above”
[[⋄ add boxes ahead to queue and continue
boxesToDo := *boxesToDo*; [*box b* “ahead left”; *d* + 1 × *m*].
if *box b* “ahead left” ≠ *box b* “ahead right”
[[*boxesToDo* := *boxesToDo*; [*box b* “ahead right”; *d* + 1 × *m*]]].

```

    nextBox]]
    else [[if ↔boxesToDo > 0 [[nextBox]]]].
old boxesToDo.

` using accel, move porter p ahead one box
new b: point box:= porter p “below”.
box:= (b; “porter”) → numporters | box. draw b “grey”.
randNext. b:= box b (randNint 0 2 = 0 ⇐ “ahead left” ⇐ “ahead right”).
if box b “porter” < numporters [[draw b “red”. stop]]. ` crash
porter:= (p; “below”) → b | porter. box:= (b; “above”) → p | box. draw b “blue”.
old b.
new speed:= sqrt (porter p “speed”^2 + 2×accel×m) ∧ speedlimit.
porter:= (p; “arrival time”) → (porter p “arrival time” + 2×m/(porter p “speed” + speed))
    | (p; “speed”) → speed
    | porter.

old speed. old accel. old p. `these olds are not really necessary
await (iterationStartTime+visualDelayTime).
infiniteLoop]] `end of simport

```

Quote Notation Lengths

` program to compare [quote notation](#) lengths with numerator/denominator lengths

`output: screen

`use: bin div even nat odd

```

new shl ((⟨n: nat. ⟨m: nat. ` shift n left m places; n×2^m
value r: nat:= n [[for i: 0;..m [[r:= r×2]]]])).

```

```

new shr ((⟨n: nat. ⟨m: nat. ` shift n right m places; floor (n×2^-m) or div n (2^m)
value r: nat:= n [[for i: 0;..m [[r:= div r 2]]]])).

```

```

new gcd ((⟨a: nat+1. ⟨b: nat+1. ` greatest common divisor of a and b
a=b ⇐ a ⇐ a<b ⇐ gcd a (b-a) ⇐ gcd (a-b) b)).

```

```

new norm [[plan num:= nat+1 [[plan denom:= nat+1 ` normalize num/denom
[[new g:= gcd num denom. num:= num/g. denom:= denom/g]]]].

```

new count: nat:= 0. ` number of examples

new qlen: nat:= 0. ` total length of quote representations

new rlen: nat:= 0. ` total length of numerator/denominator representations

for length: 1;..15

[[for string: 0;..(shl 1 length) ` each string of that length

[[for quote: 0;..length ` each quote position (at least one bit to left of quote)

[[if even (shr string (length-1)) ≠ even (shr string (quote-1)) ` roll-normalized

[[if ` repeat-normalized

value repeatnorm: bin:= ⊤

[[new len: nat:= div (length-quote) 2. ` the length of the possibly repeating part

```

trythislen [[if len>0 ` 1 ≤ len ≤ (length-quote)/2
  [[new extract ((i: nat. ⟨l: nat. ` index i length l
    shr string i – shl (shr string (i+l) l )))].
  new ex:= extract quote len.
  if ` the negative part is a repetition (twice or more) of ex
  value b: bin:= ⊤
  [[new i: nat:= quote+len. ` i+len ≤ length
    iloop [[new ey:= extract i len.
      if ex=ey [[i:= i+len. ` i≤length
        if i+len ≤ length [[iloop]]
        else [[b:= ⊥]]]
      else [[b:= ⊥]]]]]
  [[repeatnorm:= ⊥] else [[len:= len-1. trythislen]]]]]
[[for point: 0;..length+1 ` each point position (right end, interior, left end)
  [[if ` the rightmost bit is 1 or it is to the left of quote or point
    odd string ∨ (quote=0) ∨ (point=0)
  [[ ` convert to numerator/denominator
    new num: nat:= shl string (length-quote) – string – shl (shr string quote) length.
    if num<0 [[num:= -num]].
    new denom: nat:= shl (shl 1 (length-quote) – 1) point.
    norm num denom.
    ` update statistics
    count:= count+1. qlen:= qlen+length.
    rlen:= rlen+1. ` for the sign
    loop [[num:= div num 2. rlen:= rlen+1.
      if num>0 [[loop]]].
    loop [[denom:= div denom 2. rlen:= rlen+1.
      if denom>0 [[loop]]]]]]]]]]].
! “In ”; count; “ examples, quote average length = ”;
  qlen/count; “, num/denom average length = ”; rlen/count.

```

old shl. old shr. old gcd. old norm. old count. old qlen. old rlen

Minimum Redundancy Codes

new MRC ` a program to compute minimum redundancy prefix codes (Huffman codes)

```

[[ input: keys
  `output: screen
  `use: end find nat nil nl point real text

```

new forest: *[real; tree ([text], [tree; tree])]= nil. `The data structure is a string of lists.
 `Each list is a pair of real and tree; the real is the frequency of the tree.
 `Each tree is binary with texts at the leaves.

inputstart

```

[[! “Enter a frequency, then a colon, then a message, then end , and repeat. ”;

```

```

  “To end, just enter end .”; nl.

```

readloop

```

[[?!.

```

```

  if ↔? = 0 ` Just end was pressed.

```

```

[[if ↔forest = 0 `We have not had any input yet. We need at least one.
  [! “Insufficient input. Try again.”. inputstart]].
new c:= find “:” (?).
if c = ↔? [! “Bad format: no colon. Try again.”. readloop]]
new freq:= ?_(0;..c).
if freq=-∞ [! “Bad frequency format. Try again.”. readloop]].
new message:= ?_(c+1;..↔?).
` find where the new data goes in forest and put it there.
new i: nat:= 0.
findloop [[if i = ↔forest ∨ (freq ≤ (forest_i)0) ` found where it goes
  [[forest:= forest_(0;..i); [freq; [message]]; forest_(i;..↔forest)]]
  else [[i:= i+1. findloop]]. readloop]]]].

```

` forest is now a nonempty string of pairs, each pair consisting of a frequency and a tree, each
 ` tree is a single leaf, each leaf is a list-text. They are in non-decreasing frequency order.
 ` For example: [3; [“c”]]; [4; [“a”]]; [9; [“f”]]; [12; [“b”]]; [15; [“e”]]; [20; [“d”]]

new here: nat:= 0. ` A new tree must be moved to position *here* or later.

```

loop [[if ↔forest ≥ 2
  [ ` combine the first two trees into a new tree t
  new t:= [(forest_0)0 + (forest_1)0; [(forest_0)1 ; (forest_1)1]].
  ` remove those first two trees from the forest
  forest:= forest_(2;..↔forest).
  ` put tree t into its place in the forest
  innerloop [[if here = ↔forest ∨ (t 0 < (forest_here)0) ` we have found where it goes
    [[forest:= forest_(0;..here); t; forest_(here;..↔forest). loop]]
    else [[here:= here+1. innerloop]]]].

```

` forest is now a single pair consisting of the total of all frequencies and a code tree.

```

new t:= forest_1. ` the code tree
` Walk the tree, depth-first, printing leaves and their codes
new p: point t:= nil. ` a path within t starting at the root
new pt: text:= “”. ` same path as p but as a text for printing
new back [[plan p:= *nat [[p:= p_(0;..↔p-1)]]].
loop [[if ~(t p): text ` we are at a leaf
  [! “code: ”; pt; “; message: ”; ~(t p); nl]]
  else [[p:= p;0. pt:= pt;“0”. loop. back p. back pt.
    p:= p;1. pt:= pt;“1”. loop. back p. back pt]]]] `end of MRC

```

Read Password

` program to read a password, allowing corrections, displaying blobs

`input: keys

`output: screen

`use: char delete end text

new password: text:= “”.

pswd [!“Please enter password followed by end : ”.

read [? nil (char) nil.

if ?=end [[if password=“” [!“Empty password. Try again.”; nl. pswd]]

```

    else [!nl]]
else [[if ?=delete [[if password≠"" [[password:= password_(0;..↔password-1). !delete]]]
    else [[password:= password; ?. !"•"]].
read]]]]

```

Grammars

LL(1) Grammar

In this grammar, for each nonterminal, every production except possibly the last begins with a different terminal. Director sets are needed to create the parser, but they are not needed for parsing, and that is a special case of LL(1) that deserves its own name; perhaps LL($1/2$). To parse a program, the parse stack begins with only the program nonterminal on it, and ends empty with no more input. A name control program is responsible for classifying names. For efficiency, the productions (except possibly the last) for each nonterminal should be placed in order of frequency. The following nonterminals can be eliminated by replacing them with their one production: program sequent data data6 data5 data4 data3 data1. This leaves the grammar with $33-8 = 25$ nonterminals.

program	sequent moresequents
moresequents	. program empty
sequent	phrase parallelphrases
parallelphrases	sequent empty
phrase	new simplename afternewname old simplename compounder [[program]] if data [[program]] elsepart case data [[program]] morecases elsepart for simplename : data [[program]] plan simplename parameterkind [[program]] arguments ! data ? inputafterq simplename aftersimplename
afternewname	: data := data (data) := data ? data ! data [[program]] \ afterbackslash #1 simplename compounder empty
afterbackslash	simplename afternewname

	\ simplename compounder empty
compounder	\ simplename compounder empty
elsepart	else [[program] empty
morecases	[[program] morecases empty
parameterkind	: data := data ! data ? data \
aftersimplename	[[program] compounder aftername
aftername	:= data ! data ? inputafterq arguments
inputafterq	! echo data (data) data afterpattern
afterpattern	! echo empty
echo	simplename compounder empty
arguments	number arguments ∞ arguments text arguments \top arguments \perp arguments value simplename : data := data [[program] arguments { data } arguments [data] arguments (data) arguments < simplename : data . data > arguments simplename specificand arguments empty
data	data6 moredata

moredata	\models data \Rightarrow data empty
data6	data5 moredata6
moredata6	$=$ data5 moredata6 \neq data5 moredata6 $<$ data5 moredata6 $>$ data5 moredata6 \leq data5 moredata6 \geq data5 moredata6 $:$ data5 moredata6 $::$ data5 moredata6 \in data5 moredata6 empty
data5	data4 moredata5
moredata5	, data4 moredata5 ,... data4 moredata5 data4 moredata5 \triangleleft data \triangleright data4 moredata5 empty
data4	data3 moredata4
moredata4	+ data3 moredata4 – data3 moredata4 ;; data3 moredata4 ; data3 moredata4 ;.. data3 moredata4 ' data3 moredata4 empty
data3	data2 moredata3
moredata3	\times data2 moredata3 $/$ data2 moredata3 \wedge data2 moredata3 \vee data2 moredata3 empty
data2	# data2 – data2 \sim data2 + data2 \square data2 ζ data2 * data2 ϕ data2

	\$ data2 ↔ data2 data1 moredata2
moredata2	* data2 moredata2 → data2 moredata2 ^ data2 moredata2 ^^ data2 moredata2 empty
data1	data0 moredata1
moredata1	% moredata1 ? moredata1 ?? moredata1 _ data0 moredata1 @ data0 moredata1 & data0 moredata1 arguments
data0	number ∞ text ⊥ ⊥ ? ?? value simplename : data := data [program] { data } [data] (data) < simplename : data . data > simplename specificand
specificand	(data) compounder

LR(0) Grammar

The following grammar has no reduce-reduce choices and no shift-reduce choices. It has shift-shift choices. Such a grammar is commonly called LR(0), but it should not be, because a shift action pushes an input symbol onto the parse stack, and therefore a shift action depends on the input symbol. It is a special case of LR(1) that deserves its own name, but not LR(0); perhaps LR($1/2$). To parse a program, the parse stack begins empty, and ends with only the program nonterminal on it and no more input. A name control program is responsible for classifying names.

program	sequent program . sequent
sequent	phrase

sequent || phrase

phrase

new name : data := data
new name := data
new name (data)
new name [[program]]
new name ? data ! data
new name #1
new name \
new name \\
new name name
new name
old name
name := data
name ! data
! data
name ? data (data) data ! name
? data (data) data ! name
name ? ! name
name ? data (data) data !
? ! name
? data (data) data !
name ? !
? !
name ? data (data) data
? data (data) data
simplename [[program]]
if data [[program]]
if data [[program]] **else** [[program]]
case data cases
case data cases **else** [[program]]
for simplename : data [[program]]
[[program]]
plan

cases

[[program]]
cases [[program]]

plan

plan simplename : data [[program]]
plan simplename := data [[program]]
plan simplename ? data [[program]]
plan simplename ! data [[program]]
plan simplename \
[[program]]
plan data0
name

data

data6 ≡ data ⇒ data
data6

data6

data6 = data5

	$\text{data6} \neq \text{data5}$ $\text{data6} < \text{data5}$ $\text{data6} > \text{data5}$ $\text{data6} \leq \text{data5}$ $\text{data6} \geq \text{data5}$ $\text{data6} : \text{data5}$ $\text{data6} :: \text{data5}$ $\text{data6} \in \text{data5}$ data5
data5	$\text{data5} , \text{data4}$ $\text{data5} \dots \text{data4}$ $\text{data5} \text{data4}$ $\text{data5} \triangleleft \text{data} \triangleright \text{data4}$ data4
data4	$\text{data4} ; \text{data3}$ $\text{data4} ;.. \text{data3}$ $\text{data4} ;; \text{data3}$ $\text{data4} ' \text{data3}$ $\text{data4} + \text{data3}$ $\text{data4} - \text{data3}$ data3
data3	$\text{data3} \times \text{data2}$ $\text{data3} / \text{data2}$ $\text{data3} \wedge \text{data2}$ $\text{data3} \vee \text{data2}$ data2
data2	$+ \text{data2}$ $- \text{data2}$ $\notin \text{data2}$ $\$ \text{data2}$ $\leftrightarrow \text{data2}$ $\# \text{data2}$ $\sim \text{data2}$ $\square \text{data2}$ $\not\leq \text{data2}$ $* \text{data2}$ $\text{data1} * \text{data2}$ $\text{data1} \rightarrow \text{data2}$ $\text{data1} \wedge \text{data2}$ $\text{data1} \wedge \wedge \text{data2}$ data1
data1	$\text{data1} \text{ data0}$ $\text{data1} _ \text{data0}$ $\text{data1} @ \text{data0}$ $\text{data1} \% \text{data0}$

	data1 & data0
	name ?
	name ??
	?
	??
	data0
data0	number
	∞
	text
	\top
	\perp
	[data]
	{ data }
	(data)
	< simplename : data . data >
	value simplename : data := data [program]
	name
	simplename (data)
name	simplename
	name \ simplename

Acknowledgements

The first public mention of ProTem was

E.C.R.Hehner, T.S.Norvell: “ProTem: a Programming System”, University of Toronto, Computer Systems Research Group, technical report CSRG213, 1988 September

On 1991 April 16, Eric Hehner gave a talk at the University of Waterloo titled “the ProTem Programming System”.

Theo Norvell wrote an MSc thesis in 1988 titled “Expressions, Types, and Data Structures in ProTem”. Jim Horning suggested error recovery using the *session* file. Hugh Redelmeier acted as design consultant and critic in 1990. Brian Parkinson found a bug in the implementation in 1990, and he wrote an MSc thesis in 1991 titled “Automated Theorem Proving in the ProTem Programming Language”. The design of ProTem has been improved since then, and the old implementation is now out-of-date. A new [implementation](#) is partly written.